
Introduction
This is a collection of twenty-one games written to be played on your Electron. Every game is complete in itself so you can turn to whichever one takes your fancy, type it in and play it. We've tried to include something for everyone and each one has its own detailed description so that you'll know what to expect before you embark on it. You also have a chance to see what to expect as there are samples of the displays produced on your TV screen. Of course, these cannot really do justice to many of the programs which use colour graphics - and we cannot find any way of letting you hear the accompanying sound effects.

All the games are written in BASIC and, presented in this form, they serve a dual purpse. Typing in games for yourself is a good way to absorb BASIC. If you are a beginner you will soon become familiar with its syntax and structure and, if you already have some experience, you will quickly pick up some handy techniques that you can incorporate into your own programs.

What's to follow

It's really impossible to indicate the range of programs included in this book as they do not fall into neat categories. Of the twenty-one programs, about two-thirds can be described as moving graphics games. Some of these are variations on familiar favourites, for example Positron Invaders, Rainbow Squash and Bobsleigh. Others have titles that probably won't ring any bells - Sheepdog Trials, Commando Jump and Across the Ravine - but we hope they will soon become popular once you start to play them. Laser Attack and Mighty Missile are both fast-moving 'zap-the-enemy' type games with special features that make them very different from others we've played. Treasure Island is another program that is out of the ordinary. It is a game that tests your memory and relies on a variety of interesting graphics techniques. Electron Epsom presents a horse race run to a familiar and appropriate tune. Capture the Quark is a board game in which you play against the computer on an eight-by-eight grid. There are also some programs for traditional pastimes - Magic Dice, Noughts and Crosses, Word Scramble and Mirror Tile all come into this category. Smalltalker is a rather unusual program that enables your Electron to hold a conversation with you. If you think we are joking, you'll have to try it for yourself!

Improve your programming

This book isn't, however, intended as just another collection of programs. As well as hoping to provide programs that you can have hours of fun with, we also hope to cater for all Electron owners by presenting a book that can be used in more than one way. True, you can use it simply as a source of exciting games programs but, on the other hand, you can use it to further improve your knowledge of Acorn BASIC programming. Each program is accompanied by an outline of its subroutine structure, details of special programming techniques and suggestions for further improvements. These sections are included for those of you who want to develop your own programming skills. By giving away some of our 'trade secrets' we hope that you'll be able to extend your range of techniques.

It is because we would like to be able to help you experiment with your own programming that all these games stick to BASIC. This means, of course, that the games cannot be as fast-moving or as complicated as the ones that are available pre-programmed on cassettes and which are written in machine code. But, if you want to learn to write your own programs, it is far easier to start with BASIC than to attempt to come to terms with machine code.

Perfect programming

The programs included have all been extensively tested in the form in which they appear and then printed out directly from working versions. This means that if you type in exactly what is printed in this book every program should work for you every time you run it.

However, it's a well-known fact that bugs creep in whenever you enter a program - so, if a program won't work when you've typed it in, check it very carefully against the listing. If it still won't work, have a cup of tea and check again - it's all too easy to read what you think should be there rather than what is there! If there are any particular points to look for when entering a program, you'll be alerted to them in the section on Typing Tips. Common sources of possible errors are: confusing full stops and commas or semi-colons and colons, omitting brackets (or putting in extra ones), or misreading 'less than' and 'greater than' symbols (getting these round the wrong way will lead to chaotic results).

If you do get an error message when you try to RUN a program, don't just give up but use the information it gives you to trace your mistake. The error will not necessarily be in the line whose number is reported but that line will try to use some part of the program with the bug in it. If the line uses a variable or an array, check to see if it was defined properly. If the line identified goes to another part of the program or calls a procedure or a subroutine, make sure that section is complete. For more information about programming in BASIC, consult The Electron Programmer, also published by Granada.
The Electron has a number of features that you may find helpful when entering programs. The first is the automatic numbering facility that provides you with line numbers that go up in jumps of ten, which is the convention followed throughout this book. Type AUTO before you start typing in (or AUTO next line number if you want to continue a partly typed listing). Then the Electron allows you to enter keywords with a single keypress when you press the CAPS LK/FUNC key at the same time as the key with the required command on it. There is also the copying facility which makes use of the cursor control keys together with the COPY key.

Cassette tapes

Typing in long programs can be a very frustrating business. It's not easy to avoid typing mistakes altogether and there is always the risk that you'll lose your program before you've saved it after hours of careful effort. It's far too easy to disconnect your power supply and you can't guard against thunderstorms or other sources of voltage fluctuations without great expense! Many of the programs in this book are indeed long. This is unavoidable as the games concerned include lots of features. But you can avoid having to type them all in got yourselves. The programs, exactly as listed here, are available on a pair of cassette tapes. For full details and an order form, send a stamped, self-addressed envelope to:


RAMSOFT


P.O. Box 6


Richmond


North Yorkshire


DL10 4HL

Across the Ravine

Have you the skill and judgement needed to lead an expedition through dangerous terrain? This colour graphics game provides an easy way to discover your potential. The object of the game is to get your party of five intrepid explorers across a deep ravine with a fast flowing river at the bottom of steep cliffs. There is only one option, to swing across on a rope. The rope swings all the time so each man must run and leap to catch it - any one who mistimes his jump falls into the river and is lost. Listen out for the sound effects as a man falls towards the river!

How to play

This game is all a matter of timing - you have to judge when to start each run for the rope. It is vital to catch the rope on the downswing and each man can jump approximately his own width. You can wait as long as you like before making any run and when you are ready press any key to jump. When your first little man has made his run and either swung to safety or perished, the second explorer will appear in position. There are five of them in all and the length of the rope remains the same for all of them. In each new game the rope changes at random.

Typing tips

This program - like every other one in this book - has been numbered in jumps of 10. This means that rather than type in every line number, you can get the Electron to provide them by typing "AUTO" before you start entering the program. If you do not want to start at 10, for example if you are adding to an existing listing, type "AUTO next line number". Remember that you can save effort by using the Electron's keywords and by taking advantage of its copy facility when inputting lines that are similar to earlier ones. Use the cursor keys to move to the characters you want to copy and then press the COPY key.

In line 740 there are two sets of quotation marks with nothing in between. This is known as a 'null string'. Notice that there is one space between the quotes in line 610.

Subroutine structure


20
Initialises arrays


40
Main play loop


110
Swings rope


400
Calculates positions of rope


470
Prints ravine


680
Plots man on end of rope


740
Man jumps routine


860
Gets next man ready


980
Man falls in water routine


1150
Sets up game


1240
End of game

Programming details

This program uses an interesting combination of low resolution and high resolution graphics. The river banks are made up of straightforward low resolution graphics - the solid blocks defined as CHR$(224). The rope is plotted in high resolution graphics and its old positions are blanked out by replotting them. The little man figure is a low resolution user-defined graphic, CHR$(225), PRINTed in high resolution mode. This is made possible by using a VDU 5 command and means that he can appear anywhere on the screen. He therefore seems to move smoothly rather than in the jerky fashion that would result if he could only be printed at set character positions.

Another point to note is the way the co-ordinates of the positions of the swinging rope are first calculated by subroutine 400 and stored in two arrays, 'X' and 'Y'. These positions are then used repeatedly in the plotting of the swinging rope. This saves having to recalculate them each time they are needed and so speeds the whole program up. This technique can be applied to any situation where anything is moving rhythmically or periodically.

Scope for improvement
You can make the game easier by increasing the value to the right of the < in line 830 or make it more difficult by decreasing this value.

Program

   10REM Across the Ravine

   20DIM X(32)

   30DIM Y(32)

   40GOSUB 470

   50GOSUB 400

   60GOSUB 1150

   70GOSUB 860

   80GOSUB 110

   90IF MEN=0 THEN GOTO 1240

  100GOTO 80

  110FOR T=1+R TO N-R

  120GCOL 4,4

  130PLOT 4,500,950

  140PLOT 9,X(T),Y(T)

  150S=1: GOSUB 680

  160PLOT 4,500,950

  170PLOT 9,X(T),Y(T)

  180IF J=1 THEN S=2: GOSUB 680

  190NEXT T

  200IF C=1 THEN GOTO 310

  210FOR T=N-R TO 1+R STEP -1

  220GCOL 4,4

  230PLOT 4,500,950

  240PLOT 9,X(T),Y(T)

  250S=3: GOSUB 680

  260PLOT 4,500,950

  270PLOT 9,X(T),Y(T)

  280IF J=1 THEN  S=4: GOSUB 680

  290NEXT T

  300RETURN

  310ACROSS=ACROSS+1

  320DX=(ACROSS-1)*50+25

  330DY=504

  340GOSUB 710

  350C=0

  360MEN=MEN-1

  370IF MEN=0 THEN GOTO 920

  380GOSUB 860

  390RETURN

  400N=0

  410FOR T=-PI/6 TO PI/6 STEP.05

  420N=N+1

  430X(N)=-450*SIN T

  440Y(N)=-450*COS T

  450NEXT T

  460RETURN

  470MODE 1

  480VDU 19,3,3,0,0,0

  490VDU 19,0,0,0,0,0

  500VDU 19,1,2,0,0,0

  510VDU 23,224,&FF,&FF,&FF,&FF,&FF,&FF,&FF,&FF

  520VDU 23,225,&18,&18,&7E,&18,&3C,&66,&66,&00

  530CLS

  540COLOUR 1

  550FOR I=17 TO 21

  560FOR J=0 TO 31

  570IF J>10 AND J<20 THEN GOTO 600

  580PRINT TAB(J,I);CHR$(224)

  590GOTO 620

  600IF I<19 THEN GOTO 620

  610PRINT TAB(J,I);" "

  620NEXT J

  630NEXT I

  640C=0

  650J=0

  660VDU 5

  670RETURN

  680IF C=0 THEN GOTO 740

  690DY=940+Y(T)

  700DX=485+X(T)

  710MOVE DX,DY

  720PRINT CHR$(225)

  730RETURN

  740IF INKEY$(0)="" AND J=0 THEN FOR Q=1 TO 100:NEXT Q:RETURN

  750J=1

  760DY=MY

  770DX=MX

  780GOSUB 710

  790MX=MX-80

  800DY=MY

  810DX=MX

  820GOSUB 710

  830IF ABS (MX-500-X(T))<25 AND S=2 THEN C=1:GOTO 710

  840IF MX<680 THEN GOTO 980

  850RETURN

  860MY=504

  870MX=1000

  880DY=MY

  890DX=MX

  900J=0

  910GOSUB 710

  920VDU 4

  930PRINT TAB(1,0);"men left   men across          men lost"

  940PRINT TAB(4,1);MEN;TAB(13);ACROSS;TAB(23);LOST

  950IF MEN=0 THEN GOTO 1240

  960VDU 5

  970RETURN

  980GOSUB 710

  990MY=MY-20

 1000DY=MY

 1010GOSUB 710

 1020SOUND 1,-15,MY/8,4

 1030FOR Q=1 TO 50:NEXT Q

 1040GOSUB 710

 1050IF MY>350 THEN GOTO 990

 1060SOUND 1,-15,2,6

 1070LOST=LOST+1

 1080C=0

 1090J=0

 1100MEN=MEN-1

 1110IF MEN=0 THEN GOTO 920

 1120FOR Q=1 TO 500:NEXT Q

 1130GOSUB 860

 1140RETURN

 1150MEN=5

 1160LOST=0

 1170ACROSS=0

 1180J=0

 1190C=0

 1200R=RND(4)-1

 1210PRINT TAB(0,15);SPC(10)

 1220PRINT TAB(0,16);SPC(10)

 1230RETURN

 1240VDU 4

 1250PRINT TAB(0,25);"You lost ";LOST

 1260INPUT "Another game ",A$

 1270IF LEFT$(A$,1)="Y" THEN RUN

 1280VDU 20

 1290CLS

Sheepdog Trials

If you've ever watched a shepherd and his dog coax a flock of sheep into a pen, you're bound to agree that it's quite an astounding feat. The experienced shepherd makes it all look so effortless as he shouts and whistles commands to his dog, who obediently stands his ground or edges up a few paces, or runs around the back of the flock to head off a straggler.

This game is an extremely realistic simulation. In fact, it's so true-to-life that the only person we know who's scored a 'Super Shepherd' rating was a real sheep farmer! Five fluffy white sheep and a black dog appear in a green field surrounded by a picket fence - it creates just the right country atmosphere.

How to play

The object of the game is to herd all five sheep into the pen at the top right-hand of their field in the minimum number of moves. To do this you have to control the dog using the arrow keys. If the dog approaches too close to the sheep they will scatter. (They may also scatter randomly during the course of the game just to complicate matters.) In normal play, neither the dog nor the sheep are allowed to cross any fences, although when they scatter the sheep may jump out of the pen. There will always be a total of five sheep but if they crowd very close together they will appear to merge into one another.
Once you've played this game a few times you'll realise that some strategies for controlling the sheep work better than others. Beginners tend to waste moves trying to manoeuvre the dog around the back of the flock. However, to achieve the title of 'Super Shepherd' or 'Good Dog' you'll need to make every move count.

Typing tips

The hash character is used to print the fence in subroutine 120. It is produced by typing the 3 key with 'SHIFT' key held down. The only other printing feature to look out for is the single space enclosed in double quotes in lines 440 and 780. These are used to blank out the previous positions of the dog and the sheep respectively.

Subroutine structure


20
Sets up graphics characters and arrays


120
Prints fences


270
Prints sheep


330
Prints dog


370
Moves dog


520
Moves sheep and checks for end of game


550
Move logic for sheep


780
Prints sheep


840
Prints messages and end of game


990
Scatters sheep

Programming details

Line 20 changes the cursor keys so that they return ASCII codes, thereby allowing them to be used as arrow keys in this game. At the end of the game their normal function is restored in line 960. The other *FX command used in this program (*FX 15,1 in line 380) has the effect of clearing the Electron's type ahead buffer - in other words it gets rid of any accumulation of keypresses that have not already been acted upon. Line 30 sets Mode 1 and disables the flashing cursor.

When you RUN this game you may imagine that there are some special techniques involved to govern the movement of the sheep and sheepdog. This is, however, not the case and the program depends entirely on calculating their positions relative to each other according to mathematical equations. For example, lines 600 and 610 check to see if the dog has approached too close to the sheep. If he has (or if the random number generating is less than .01, a one-in-a-hundred chance occurrence) then the sheep scatter according to the equations in 1000 and 1030. IF statements are also used to make sure that the dog does not move into any of the picket fences or that the sheep do not move on to the dog or on to the fences.

Scope for improvement
If you get really proficient at this game you can try to make it more difficult. You might increase the chance of the sheep scattering at random, by altering the value of the cut-off point for the random number in line 610, or you could add some obstacles such as a pond or a river that the sheep had to avoid or cross. Another suggestion is to modify the game to employ a time criterion, using the Electron's timer, instead of counting the number of moves needed.

Program

   10REM Sheepdog Trial

   20*FX 4,1

   30MODE 1: VDU 23,1;0;0;0;0

   40VDU 23,224,&00,&00,&7A,&FF,&7D,&78,&48,&48

   50VDU 23,225,&00,&00,&06,&7B,&78,&84,&42,&00

   60VDU 19,0,2,0,0,0

   70VDU 19,3,0,0,0,0

   80VDU 19,1,7,0,0,0

   90DIM Y(5)

  100DIM X(5)

  110M=0

  120FOR X=0 TO 15

  130PRINT TAB(X,16);"*"

  140NEXT

  150FOR Y=0 TO 16

  160PRINT TAB(16,Y);"*"

  170NEXT

  180FOR Y=0 TO 20

  190PRINT TAB(0,Y);"*";TAB(31,Y);"*"

  200NEXT

  210FOR X=0 TO 31

  220PRINT TAB(X,0);"*";TAB(X,21);"*"

  230NEXT

  240FOR Y=1 TO 3

  250PRINT TAB(12,Y);"*"

  260NEXT

  270COLOUR 1

  280FOR S=1 TO 5

  290Y(S)=5+RND(10)

  300X(S)=4+RND(6)

  310PRINT TAB(X(S),Y(S));CHR$(224)

  320NEXT S

  330COLOUR 3

  340YD=1+RND(3)

  350XD=1+RND(3)

  360PRINT TAB(XD,YD);CHR$(225)

  370COLOUR 3

  380*FX 15,1

  390D=INKEY(0)

  400IF D=-1 THEN GOTO 390

  410IF XD=12 AND YD=4 AND D=&8B THEN GOTO 390

  420IF XD=11 AND YD<4 AND D=&89 THEN GOTO 390

  430IF XD=13 AND YD<4 AND D=&88 THEN GOTO 390

  440PRINT TAB(XD,YD);" "

  450IF D=&88 AND XD>1 THEN XD=XD-1

  460IF D=&89 AND XD<15 THEN XD=XD+1

  470IF D=&8A AND YD<15 THEN YD=YD+1

  480IF D=&8B AND YD>1 THEN YD=YD-1

  490PRINT TAB(XD,YD);CHR$(225)

  500M=M+1

  510PRINT TAB(20,10);"MOVE ";M

  520GOSUB 550

  530IF F=0 THEN GOTO 840

  540GOTO 370

  550F=0

  560COLOUR 1

  570FOR S=1 TO 5

  580Y=Y(S)

  590X=X(S)

  600IF(ABS(X(S)-XD)<2 AND ABS(Y(S)-YD)<2) THEN GOSUB 990

  610IF RND(1)<.01 THEN GOSUB 990

  620IF ABS(X(S)-XD)>2+RND(2) THEN GOTO 780

  630IF ABS(Y(S)-YD)>2+RND(2) THEN GOTO 780

  640X(S)=X(S)+SGN(X(S)-XD)

  650Y(S)=Y(S)+SGN(Y(S)-YD)

  660IF X(S)<13 AND X(S)>11 AND Y(S)<4 THEN X(S)=X

  670FS=0

  680FOR Z=1 TO 5

  690IF Z=S THEN GOTO 710

  700IF (X(S)=X(Z)) AND (Y(S)=Y(Z)) THEN FS=1

  710NEXT Z

  720IF FS=1 THEN GOTO 640

  730IF X(S)=XD AND Y(S)=YD THEN GOTO 640

  740IF X(S)<1 THEN X(S)=1

  750IF X(S)>15 THEN X(S)=15

  760IF Y(S)<1 THEN Y(S)=1

  770IF Y(S)>15 THEN Y(S)=15

  780PRINT TAB(X,Y);" "

  790PRINT TAB(X(S),Y(S));CHR$(224)

  800IF X(S)>12 AND (Y(S)>0 AND Y(S)<4) THEN GOTO 820

  810F=1

  820NEXT S

  830RETURN

  840REM ENDGAME

  850COLOUR 1

  860PRINT TAB(2,22);

  870IF M<40 THEN PRINT "SUPER SHEPHERD!!":GOTO 920

  880IF M<60 THEN PRINT "GOOD DOG!!":GOTO 920

  890IF M<90 THEN PRINT "KEEP PRACTISING":GOTO 920

  900IF M<120 THEN PRINT "BETTER LUCK NEXT TIME":GOTO 920

  910PRINT "HAND IN YOUR CROOK !!"

  920PRINT TAB(2,23);"YOU TOOK  ";M;" MOVES"

  930INPUT "ANOTHER GAME Y/N ",A$

  940A$=LEFT$(A$,1)

  950IF A$="Y" THEN RUN

  960*FX 4,0

  970VDU 20

  980END

  990XT=X(S):YT=Y(S)

 1000X(S)=X(S)+(SGN(RND(1)-.5)*(2+RND(2)))

 1010IF X(S)<1 THEN X(S)=1

 1020IF X(S)>15 THEN X(S)=15

 1030Y(S)=Y(S)+(SGN(RND(1)-.5)*(2+RND(2)))

 1040IF Y(S)<1 THEN Y(S)=1

 1050IF Y(S)>15 THEN Y(S)=15

 1060IF X(S)=12 AND Y(S)<4 THEN GOTO 1000

 1070IF XT=X(S) AND YT=Y(S) THEN GOTO 1000

 1080RETURN

Laser Attack

This is a very exciting and fast moving space fight game that uses sound effects and some rather unusual graphics techniques to good effect. The screen is treated as if it were a spherical universe. So, if you go off at the right you re-appear at the left. This game is a race against the clock. You have a hundred seconds in which to annihilate the enemy ship with your infallible laser weapon. The chase is on!

How to play

At the beginning of this game you have to select a difficulty factor. This governs the unpredictability of the enemy ship's course and the number of stars that appear. The stars act as obstacles in this game. If you hit one you will be deflected at random so the fewer there are the easier it is to steer your course. Your ship moves continuously. It is shaped like an arrow-head and can point in any of eight directions.

Every time you press any key it turns clockwise through 45 degrees. The enemy is a revolving cartwheel-shaped disc that meanders through space. To fire your laser, press the up cursor key. Your weapon will fire in a straight line from the point of your arrow. If you hit the enemy ship, it will disintegrate with appropriate sound and visual effects. The time taken is constantly displayed at the top left of the screen and when it reaches 100 your time is up.

Typing tips

You may find the COPY key useful when entering some sections of this program where there are many similar lines. Notice the null string in line 150 - there should not be a space between the two sets of quote marks.

Subroutine structure


20 
Initialisation


50
Main play loop


120
Fires or rotates direction of arrow


200
Laser zap and hit logic


490
Detects hit


560
Moves arrow


700
Moves target


850
Title frame


950
Gets and prints time


990
Prints stars


1030
Initialises variables and defines graphics


1890
End of game

Programming details

This is a complicated program and one that illustrates a number of novel programming methods. Notice, for example, the way the revolving cartwheel is produced by using two user-defined graphics characters, one a version of the other rotated through 90 degrees, which are printed alternately. The way that eight versions of the arrow graphic are used in order to allow it to be moved in any of eight different directions is another interesting technique. While the player's ship, the target and the starts (asterisks) are all PRINTed in low resolution graphics, the laser zap and the explosion are PLOTted in high resolution graphics. The way the direction of movement and velocity is set in PROCINIT using the arrays W and V is also worth attention. Line 1170 disables the usual blinking cursor which would otherwise be very distracting. This is a technique you will find very useful when writing your own games.

Scope for improvement
If you feel very ambitious you could make this game even more exciting by enabling the enemy ship to fire at random - so that you have to dodge its fire at the same time as pursuing it.
Program

   10REM Laser Attack

   20MODE 1

   30PROCTITLE

   40PROCINIT

   50PROCTIME

   60IF INT(T)>1000 THEN GOTO 1890

   70PROCMOVE

   80PROCSHIP

   90IF H=1 THEN GOTO 1890

  100PROCENEMY

  110GOTO 50

  120DEF PROCMOVE

  130A$=INKEY$(0)

  140*FX 15,1

  150IF A$="" THEN ENDPROC

  160IF ASC(A$)=139 THEN GOTO 200

  170K=K+1

  180IF K>8 THEN K=1

  190ENDPROC

  200XL=X*32+16

  210YL=1023-Y*32-16

  220H=FNHIT

  230MOVE XL,YL

  240DX=0

  250IF V(K)=1 THEN DX=1279-XL

  260IF V(K)=-1 THEN DX=-XL

  270DY=0

  280IF W(K)=1 THEN DY=-YL

  290IF W(K)=-1 THEN DY=1023-YL

  300IF V(K)*W(K)=0 THEN GOTO 330

  310IF ABS(DX)<ABS(DY) THEN DY=ABS(DX)*SGN(DY):GOTO 330

  320DX=ABS(DY)*SGN(DX)

  330PLOT 1,DX,DY

  340MOVE XL,YL

  350SOUND &11,1,60,5

  360PLOT 2,DX,DY

  370IF H=0 THEN ENDPROC

  380MX=B*32+16: MY=1023-A*32-16

  390FOR I=1 TO RND(5)+20

  400DX=50-RND(100)

  410IF MX+DX>1279 OR MX+DX<0 THEN GOTO 470

  420DY=50-RND(100)

  430IF MY+DY>1023 OR MY+DY<0 THEN GOTO 470

  440MOVE MX,MY

  450PLOT 1,DX,DY

  460SOUND 0,-15,5,4

  470NEXT I

  480ENDPROC

  490DEF FNHIT

  500DY=A-Y

  510DX=B-X

  520IF W(K)*DX<>V(K)*DY THEN =0

  530IF ABS(V(K))*SGN(DX)<>V(K) OR ABS(W(K))*SGN(DY)<>W(K) THEN =0

  540REM HIT!

  550=1

  560DEF PROCSHIP

  570IF NB1=0 THEN PRINT TAB(X,Y);" "

  580X=X+V(K)

  590Y=Y+W(K)

  600IF X<0 THEN X=39

  610IF X>39 THEN X=0

  620IF Y<0 THEN Y=29

  630IF Y>29 THEN Y=0

  640IF POINT(X*32+16,1023-Y*32-16)<>0 THEN SOUND 1,-15,0,5:GOTO 680

  650PRINT TAB(X,Y);M$(K)

  660NB1=0

  670ENDPROC

  680NB1=1:K=K+1:IF K>8 THEN K=1

  690ENDPROC

  700DEF PROCENEMY

  710IF RND(1)>1.05-DF/20 THEN Z=Z+1

  720IF Z>8 THEN Z=1

  730IF NB2=0 THEN PRINT TAB(B,A);" "

  740A=A+V(Z)

  750B=B+W(Z)

  760IF B<0 THEN B=39

  770IF B>39 THEN B=0

  780IF A<0 THEN A=29

  790IF A>29  THEN A=0

  800IF POINT(B*32+16,1023-A*32-16)<>0 THEN SOUND 1,-15,0,5:NB2=1:Z=Z+1:GOTO 720

  810PRINT TAB(B,A);W$(R+1)

  820R=NOT R

  830NB2=0

  840ENDPROC

  850DEF PROCTITLE

  860PRINT TAB(8,2);"L a s e r   A t t a c k"

  870PRINT TAB(0,8);"You are in control of an advanced laser"

  880PRINT "attack ship in pursuit of an enemy craft"

  890PRINT ''"Shoot it down before your time is up !!!"

  900PRINT TAB(0,20);

  910INPUT "Select the difficulty level 1 (easy) "'"to 10 (difficult) ",DF

  920IF DF<1 OR DF>10 THEN GOTO 910

  930CLS

  940ENDPROC

  950DEF PROCTIME

  960T=TIME/10

  970PRINT TAB(0,1);INT(T/10);"  "

  980ENDPROC

  990DEF PROCSTAR

 1000IF RND(1)>.1+DF/50 THEN ENDPROC

 1010PRINT TAB(RND(39),RND(30));"*"

 1020ENDPROC

 1030DEF PROCINIT

 1040DIM W(8),V(8)

 1050DIM A(8),B(8),C(8),D(8),E(8),F(8)

 1060DIM W$(2),M$(8)

 1070DIM S$(8),R$(8)

 1080VDU 19,0,4,0,0,0

 1090VDU 19,1,3,0,0,0

 1100VDU 19,2,2,0,0,0

 1110VDU 19,3,7,0,0,0

 1120COLOUR 1:COLOUR 128

 1130DATA 0,1,1,1,1,0,1,-1,0,-1,-1,-1,-1,0,-1,1

 1140FOR I=1 TO 8

 1150READ W(I),V(I)

 1160NEXT I

 1170VDU 23,1;0;0;0;0

 1180K=1

 1190X=20

 1200Y=10

 1210V=V(K)

 1220W=W(K)

 1230S$(1)="00011000"

 1240S$(2)="00111100"

 1250S$(3)="01111110"

 1260S$(4)="11111111"

 1270S$(5)="00111100"

 1280S$(6)="00111100"

 1290S$(7)="00111100"

 1300S$(8)="00111100"

 1310R$(1)="11111000"

 1320R$(2)="11110000"

 1330R$(3)="11111000"

 1340R$(4)="11111100"

 1350R$(5)="10111110"

 1360R$(6)="00011111"

 1370R$(7)="00001110"

 1380R$(8)="00000100"

 1390FOR I=1 TO 8

 1400FOR J=1 TO 8

 1410A(I)=A(I)*2+VAL(MID$(S$(I),J,1))

 1420B(I)=B(I)*2+VAL(MID$(S$(9-J),I,1))

 1430C(I)=C(I)*2+VAL(MID$(S$(J),I,1))

 1440D(I)=D(I)*2+VAL(MID$(R$(I),J,1))

 1450E(I)=E(I)*2+VAL(MID$(R$(9-J),I,1))

 1460F(I)=F(I)*2+VAL(MID$(R$(9-I),9-J,1))

 1470PROCSTAR

 1480NEXT J

 1490NEXT I

 1500VDU 23,224

 1510FOR I=1 TO 8:VDU A(I):NEXT I

 1520VDU 23,225

 1530FOR I=1 TO 8:VDU A(9-I):NEXT I

 1540VDU 23,226

 1550FOR I=1 TO 8:VDU B(I):NEXT I

 1560VDU 23,227

 1570FOR I=1 TO 8:VDU C(I):NEXT I

 1580VDU 23,228

 1590FOR I=1 TO 8:VDU D(I):NEXT I

 1600VDU 23,229

 1610FOR I=1 TO 8:VDU D(9-I):NEXT I

 1620VDU 23,230

 1630FOR I=1 TO 8:VDU E(I):NEXT I

 1640VDU 23,231

 1650FOR I=1 TO 8:VDU F(I):NEXT I

 1660M$(1)=CHR$(226)

 1670M$(2)=CHR$(231)

 1680M$(3)=CHR$(225)

 1690M$(4)=CHR$(229)

 1700M$(5)=CHR$(227)

 1710M$(6)=CHR$(228)

 1720M$(7)=CHR$(224)

 1730M$(8)=CHR$(230)

 1740A=10:B=10

 1750Z=RND(8)

 1760VDU 23,232,&20,&42,&25,&18,&18,&A4,&42,&04

 1770VDU 23,233,&0E,&88,&88,&F8,&1F,&09,&09,&38

 1780W$(0)=CHR$(232)

 1790W$(1)=CHR$(233)

 1800R=0

 1810NB1=0

 1820NB2=0

 1830TIME=0

 1840H=0:*FX 4,1

 1850COLOUR 2

 1860GCOL 0,3

 1870ENVELOPE 1,1,100,0,-10,100,50,100,126,0,0,-126,126,126

 1880ENDPROC

 1890IF H<>1 THEN GOTO 1920

 1900PRINT TAB(0,30);"You did it !!!"

 1910GOTO 1930

 1920PRINT TAB(0,30);"Your time is up"

 1930*FX 15,1

 1940INPUT "Another game Y/N ",A$

 1950IF LEFT$(A$,1)="Y" THEN RUN

 1960CLS

4

Word Scramble

This is a game that all the family can play - and it really presents quite a challenge even to the most sharp-eyed and keen-witted. Your Electron invites you to give it a list of up to ten words, each up to ten letters long. Once you have entered them it hides them within a fifteen-by-fifteen grid and fills up all the vacant spaces with random letters. All the words you've entered appear along straight lines - vertically, horizontally or diagonally - but they can be backwards, upside-down, or slanting from bottom to top. Once they have been camouflaged by all the random letters, spotting them is like looking for a needle in a haystack. If you want to make the task a little easier, you can opt to preview the puzzle before any extra letters are added. This at least gives you a chance to unscramble the puzzle. There is yet another helpful hint you can opt for. You can have the list of hidden words displayed on the screen beside the puzzle - but you may be surprised how difficult to spot they still are.

The object of the game is to find all the hidden words. Your position in the square is indicated by an inverted character which you can use as a pointer to identify the first letter of each word you find. When you think you have found the first letter of a word, type a lower-case "w". If you are correct you score a point, otherwise you'll hear a dismal-sounding tone.

How to play

The Electron guides you through the early stages of this game, asking you first how many words you wish to supply and then prompting you for each. You must use lower-case only when supplying the words of your choice. Then it has to create the puzzle - which takes a little time - and the time is longer the more long words you've included. It tells you that it's 'working' on it so that you don't think it's forgotten about you. When it's ready asks if you want to preview the puzzle. If you prefer to play the game without any advantages you can skip the preview by answering "n". Similarly, you can answer either "y" or "n" to the next question which gives the option of having the list of hidden words displayed on the screen beside the completed grid. When the grid appears, you'll see that the top left-hand position is displayed in inverted graphics. This is where the pointer starts. You have to use the arrow keys to move this cursor to a letter that you think is the first letter of one of the words in the list. Once the cursor is in position, type "w" - again using lower-case. The Electron will then ask you for the word that you have identified. Type this in. If you are correct you will score one point (and your score total will go up by one) but if you are wrong you will hear a tone. Once you've completed the puzzle you'll see the message "You got them all". If you want to give up during the game type "r" and you'll be given the option to resign.

Typing tips

When playing this game, remember to use lower-case letters only. There is a single space between double quotes in lines 420, 600, 800, 810, 930, 1400 and 1420 so don't type in a null string instead. A null string (two sets of double quotes without any space between) does, however, occur in line 1090.

Subroutine structure


20
Initialisation routine


80
Calls finding words routine


90
Asks for words to be input


280
Gives error message if more than 10 letters input


330
Finds longest word left in list


380
Constructs puzzle


760
Prints puzzle


870
Allows preview and fills up puzzle with random letters


1010
Main play loop


1240
Asks for guess of word


1320
Word correct routine


1450
Checks that word is in list


1500
Checks for word


1620
Defines dot character and sets score to zero


1670
Resign routine


1720
End of game

Programming details

Some interesting techniques are used in this program. The words are fitted into the square by choosing starting points at random (lines 510-520) and also by choosing directions at random (lines 530-540). Each word is then tested against the square position-by-position and if there is an empty space - or the identical letter is already present - for every letter of the word then the word goes in. This allows for two or more words to cross over sharing a space at a common letter. If the word can't be fitted in at its first random spot and direction, the program jumps back and chooses another spot and direction. This procedure is repeated until all the words are fitted.

Scope for improvement
If you have a printer, you could add a routine to this program to enable the completed puzzle to be printed out so that you take it away to be solved. To make the game more difficult you could allow it to accept more words. Notice, however, that the more words there are the longer it will take to be set up initially. To make the game easier you could remove words from the word list, or mark them in some way, once they have been found.

Program

   10REM Word Scramble

   20MODE 6

   30*FX 4,1

   40PROCWORDS

   50PROCFIT

   60PROCDOT

   70PROCPREVIEW

   80GOSUB 1010

   90DEF PROCWORDS

  100DIM L$(10)

  110DIM C(10)

  120LST=0

  130CLS

  140PRINT TAB(5,2);

  150INPUT "How many words ",W

  160IF W<2 OR W>10 THEN SOUND 1,-15,50,4:GOTO 140

  170PRINT '"Enter words (lower case only)"

  180FOR I=1 TO W

  190PRINT TAB(5,5+I);"Word Number ";I;"=";

  200INPUT W$

  210IF LEN(W$)>10 THEN PROCERROR

  220IF LEN(W$)<1 THEN GOTO 200

  230L$(I)=W$

  240C(I)=LEN(W$)

  250NEXT I

  260VDU 23,1;0;0;0;0

  270ENDPROC

  280DEF PROCERROR

  290PRINT TAB(5,5+I);"maximum of ten letters!"

  300INPUT W$

  310PRINT TAB(5,5+I);"Word Number ";I;"=";W$;SPC(10)

  320ENDPROC

  330M=0: J=0

  340FOR Z=1 TO W

  350IF M<C(Z) THEN M=C(Z): J=Z

  360NEXT Z

  370RETURN

  380DEF PROCFIT

  390DIM D$(15,15)

  400FOR J=0 TO 14

  410FOR I=0 TO 14

  420D$(I,J)=" "

  430NEXT I

  440NEXT J

  450CLS

  460F=0

  470FOR I=1 TO W

  480GOSUB 330

  490L=C(J)

  500C(J)=0

  510X=RND(15-L)

  520Y=RND(15-L)

  530V=RND(3)-2

  540U=RND(3)-2

  550IF U=0 AND V=0 THEN GOTO 530

  560A=X: B=Y

  570IF V<0 THEN A=A+L

  580IF U<0 THEN B=B+L

  590FOR K=1 TO L

  600IF D$(A,B)<>" " AND D$(A,B)<>MID$((L$(J)),K,1) THEN F=1:K=L:GOTO 630

  610A=A+V

  620B=B+U

  630NEXT K

  640IF F=1 THEN F=0:GOTO 510

  650PRINT "Working"

  660A=X :B=Y

  670IF V<0 THEN A=A+L

  680IF U<0 THEN B=B+L

  690FOR K=1 TO L

  700D$(A,B)=MID$(L$(J),K,1)

  710A=A+V

  720B=B+U

  730NEXT K

  740NEXT I

  750ENDPROC

  760DEF PROCPRINT

  770CLS

  780FOR M=0 TO 14

  790FOR N=0 TO 14

  800IF D$(N,M)=" " THEN PRINT CHR$(224);

  810IF D$(N,M)<>" " THEN PRINT D$(N,M);

  820NEXT N

  830IF LST=0 OR M>10 THEN PRINT

  840IF LST=1 AND M<=10 THEN PRINT TAB(20);L$(M)

  850NEXT M

  860ENDPROC

  870DEF PROCPREVIEW

  880INPUT "Do you want to preview"'" the puzzle y/n ",A$

  890IF LEN(A$)=0 THEN GOTO 880

  900IF LEFT$(A$,1)"y" THEN PROCPRINT

  910FOR I=0 TO 14

  920FOR J=0 TO 14

  930IF D$(J,I)=" " THEN   D$(J,I)=CHR$(RND(25)+97)

  940NEXT J

  950NEXT I

  960INPUT "Do you want to display"'" the words beside the puzzle y/n ",A$

  970IF LEN(A$)=0 THEN GOTO 960

  980IF LEFT$(A$,1)="y" THEN LST=1

  990PROCPRINT

 1000ENDPROC

 1010X=0

 1020Y=0

 1030COLOUR 0

 1040COLOUR 135

 1050PRINT TAB(X,Y);D$(X,Y);

 1060COLOUR 7

 1070COLOUR 128

 1080A$=INKEY$(0)

 1090IF A$="" THEN GOTO 1080

 1100IF A$="w" THEN GOTO 1230

 1110PRINT TAB(X,Y);D$(X,Y);

 1120IF ASC(A$)=&88 AND X>0 THEN X=X-1

 1130IF ASC(A$)=&89 AND X<14 THEN X=X+1

 1140IF ASC(A$)=&8B AND Y>0 THEN Y=Y-1

 1150IF ASC(A$)=&8A AND Y<14 THEN Y=Y+1

 1160IF A$="r" THEN GOSUB 1670

 1170COLOUR 0

 1180COLOUR 135

 1190PRINT TAB(X,Y);D$(X,Y);

 1200COLOUR 7

 1210COLOUR 128

 1220GOTO 1080

 1230PRINT TAB(0,20);

 1240INPUT "What is the word ",W$

 1250IF LEN(W$)=0 OR LEN(W$)>15 THEN SOUND 1,-15,50,2: GOTO 1230

 1260GOSUB 1450

 1270PRINT TAB(0,20);SPC(30)

 1280IF MATCH=0 THEN SOUND 1,-15,50,2:GOTO 1080

 1290FOR U=-1 TO 1

 1300FOR V=-1 TO 1

 1310IF U=0 AND V=0 THEN GOTO 1340

 1320PROCMATCH

 1330IF MATCH=1 THEN V=1:U=1:GOTO 1340

 1340NEXT V

 1350NEXT U

 1360IF MATCH=1 THEN GOTO 1390

 1370SOUND 1,-15,50,4

 1380GOTO 1080

 1390SCORE=SCORE+1

 1400PRINT TAB(10,21);"Score= ";SCORE;" "

 1410SOUND 1,-15,50,4

 1420L$(WORD)=" "

 1430IF SCORE=W THEN GOTO 1780

 1440GOTO 1080

 1450MATCH=0

 1460FOR I=1 TO W

 1470IF W$=L$(I) THEN MATCH=1: WORD=I:I=W

 1480NEXT I

 1490RETURN

 1500DEF PROCMATCH

 1510MATCH=1

 1520A=X

 1530B=Y

 1540FOR I=1 TO LEN(W$)

 1550IF MID$(W$,I,1)<>D$(A,B) THEN I=LEN(W$):MATCH=0:GOTO 1600

 1560A=A+U

 1570B=B+V

 1580IF A<1 OR A>15 THEN I=LEN(W$):MATCH=0:GOTO 1600

 1590IF B<1 OR B>15 THEN I=LEN(W$):MATCH=0:GOTO 1600

 1600NEXT I

 1610ENDPROC

 1620DEF PROCDOT

 1630REM dot character

 1640VDU 23,224,&00,&00,&00,&18,&18,&00,&00,&00

 1650SCORE=0

 1660ENDPROC

 1670PRINT TAB(0,22);

 1680INPUT "Are you sure that you"'"can find no more words y/n ",A$

 1690PRINT TAB(0,22);SPC(70)

 1700IF LEFT$(A$,1)<>"y" THEN RETURN

 1710CLS

 1720PRINT TAB(10,8);"Final score= ";SCORE

 1730PRINT TAB(8,10);

 1740INPUT "Another game y/n ",A$

 1750IF A$="y" THEN RUN

 1760IF A$<>"n" THEN GOTO 1730

 1770END

 1780PRINT TAB(0,22);"You got them all!"

 1790GOTO 1740

>*SPOOL

N RETURN

 1710CLS

 1720PRINT TAB(10,8);"Final score= ";SCORE

 1730PRINT TAB(8,10);

 1740INPUT "Another game y/n ",A$

 1750IF A$="y" THEN RUN

 1760IF A$<>"n" THEN GOTO 1730

 1770END

 1780PRINT TAB(0,22);"You got them all!"

 1790GOTO 1740

Treasure Island

Find the hidden treasure before the pirate ship reaches the island. This game has all the ingredients of high adventure. A desert island, peopled by natives both hostile and friendly, gold buried at the spot marked 'X' on the map, quicksands that spell danger to the unlucky treasure-seeker and even Long John Silver's parrot. While you hunt for the booty, the pirate ship is heading towards the island and once the pirates land you are certain to be captured. The game is displayed in colour graphics and has sound effects as well.
How to play

The treasure is buried at the spot marked 'X' on the map that is briefly flashed on to the screen at the start of the game. The path is also indicated. You have to follow the path exactly. If you stray, there are three possible outcomes. If you are lucky, Long John Silver's parrot will guide you back to the path - you will see the parrot hovering over the next position on the path. If you are unlucky you will encounter hostile natives and will find yourself back on the path three paces back from where you left it. If you are unluckier still you will end up in a quicksand. This can be a final fate or you may be rescued by a friendly native. If you need to consult the map in order to follow the path you can type "h". When you do this you will be shown the map - now also indicating the locations of the quicksands - for a short, random length of time. But every time you ask to see the map the pirate ship comes nearer, and if the ship arrives before you find the treasure you will be captured. The ship advances, anyway, once for every five moves you make so you need to be accurate. To move along the path, use the right, left and forward arrow keys - you cannot move backwards.

Subroutine structure


20
Sets up screen display


90
Defines graphics characters


230
Initialises variables and arrays


310
Main play loop


510
Logic for natives' attack


670
Logic for parrot's help


810
Prints map


1030
Moves man


1250
Prints quicksand on map


1330
Logic for quicksands


1460
Constructs island


1590
Constructs path


1680
Locates treasure


1710
Constructs quicksand


1800
Moves, prints pirate ship


1950
Prints island


2150
Help routine


2220
Treasure found routine


2400
End of game

Programming details

This is a very long program with lots of different ingredients. It therefore appears rather complicated whereas, in fact, it is quite straightforward. One technique to notice is the way the island is constructed at random by subroutine 1460. There the user of SGN function results in the contour of the island first going out and then coming in at random, giving an island-like shape.

Scope for improvement
If you want to alter the length of time the map is displayed for when you ask to see it, you can change the final value of the FOR loop in line 2170. You might also want to alter the amount that the pirate ship moves at each step by setting a different value for 'R' in line 1840. Currently it's set randomly to a value between 1 and 3.

Program

   10REM Treasure Island

   20MODE 1

   30VDU 23,1;0;0;0;0

   40VDU 19,0,4,0,0,0

   50VDU 19,1,2,0,0,0

   60VDU 19,2,3,0,0,0

   70VDU 19,3,0,0,0,0

   80*FX 4,1

   90REM parrot

  100VDU 23,224,&20,&36,&3E,&1C,&1E,&27,&40,&00

  110REM native

  120VDU 23,225,&01,&19,&D9,&FF,&D9,&19,&25,&25

  130REM ship sail

  140VDU 23,226,&FF,&7E,&D5,&C3,&C3,&D5,&7E,&FF

  150REM ship

  160VDU 23,227,&10,&FF,&FF,&7E,&7E,&73,&3C,&3C

  170REM quicksand

  180VDU 23,228,&0C,&1C,&3E,&7F,&FF,&FE,&7C,&38

  190REM man

  200VDU 23,229,&18,&18,&7E,&18,&3C,&66,&66,&00

  210REM island

  220VDU 23,230,&FF,&FF,&FF,&FF,&FF,&FF,&FF,&FF

  230F=0

  240XS=1: YS=1

  250MES=0

  260DIM V(10)

  270DIM U(10)

  280DIM L(20)

  290DIM R(20)

  300DIM X(20)

  310GOSUB 1460

  320XM=X(T+1)

  330YM=T+1

  340GOSUB 810

  350FOR Q=1 TO 2000+RND(1000):NEXT Q

  360GOSUB 1950

  370GOSUB 1160

  380GOSUB 1040

  390IF XM=XT AND YM=YT THEN GOTO 2220

  400F=F+1

  410IF INT (F/5)=F/5 THEN GOSUB 1800

  420IF X(YM)=XM AND INT (F/5)=F/5 THEN GOTO 360

  430IF X(YM)=XM THEN GOTO 380

  440REM off path

  450SOUND 0,-15,10,2:GOSUB 1800

  460FOR Q=1 TO 10

  470REM test for quicksands

  480IF V(Q)=XM AND U(Q)=YM THEN GOSUB 1330

  490NEXT Q

  500IF RND(1)<=.4 THEN GOTO 670

  510REM natives logic

  520GOSUB 1950

  530PRINT TAB(1,25);"HOSTILE NATIVES AHEAD"

  540FOR N=1 TO 3

  550R=RND(3)

  560IF YM+R>=B THEN R=0

  570COLOUR 129

  580COLOUR 3

  590PRINT TAB(XM,YM+R);CHR$(225)

  600NEXT N

  610YM=YM-3

  620IF YM<=T+1 THEN YM=T+1

  630XM=X(YM)

  640PRINT TAB(XM,YM);CHR$(229)

  650MES=1

  660GOTO 380

  670REM parrot logic

  680GOSUB 1950

  690GOSUB 1160

  700COLOUR 128

  710COLOUR 3

  720PRINT TAB(1,25);"FOLLOW LONG JOHN SILVERS PARROT"

  730YJ=YM+1

  740IF YJ>P THEN YJ=P

  750XJ=X(YJ)

  760COLOUR 129

  770COLOUR 3

  780PRINT TAB(XJ,YJ);CHR$(224)

  790MES=1

  800GOTO 380

  810REM print island

  820COLOUR 130

  830CLS

  840COLOUR 3

  850FOR X=L(T) TO R(T)

  860PRINT TAB(X,Y);CHR$(230)

  870NEXT X

  880FOR Y=T TO B

  890PRINT TAB(L(Y),Y);CHR$(230);TAB(R(Y),Y);CHR$(230)

  900NEXT Y

  910FOR X=L(Y-1) TO R(Y-1)

  920PRINT TAB(X,Y);CHR$(230)

  930NEXT X

  940REM print path

  950FOR Y=T+1 TO  P

  960PRINT TAB(X(Y),Y);"*"

  970NEXT Y

  980PRINT TAB(1,20);"X MARKS THE SPOT"

  990PRINT TAB(X(P),P);"X"

 1000PRINT TAB(XM,YM);CHR$(229)

 1010GOSUB 1250

 1020RETURN

 1030REM get input and

 1040REM move man

 1050SOUND 1,-15,50,2

 1060A$=INKEY$(0)

 1070IF A$="" THEN GOTO 1060

 1080COLOUR 129

 1090COLOUR 3

 1100PRINT TAB(XM,YM);" "

 1110IF A$="H" THEN GOSUB 2150: RETURN

 1120IF ASC(A$)=&88 THEN XM=XM-1: GOTO 1150

 1130IF ASC(A$)=&89 THEN XM=XM+1: GOTO 1150

 1140IF ASC(A$)<>&8A THEN GOTO 1050

 1150YM=YM+1

 1160COLOUR 129

 1170COLOUR 3

 1180PRINT TAB(XM,YM);CHR$(229)

 1190IF MES=0 THEN RETURN

 1200COLOUR 128

 1210FOR I=0 TO 31

 1220PRINT TAB(I,25);" "

 1230NEXT I

 1240MES=0: RETURN

 1250REM print quicksands

 1260COLOUR 3

 1270FOR Q=1 TO 10

 1280PRINT TAB(V(Q),U(Q));CHR$(228)

 1290NEXT Q

 1300XS=XS+1

 1310YS=YS+1

 1320RETURN

 1330REM quicksand

 1340GOSUB 1950

 1350COLOUR 2

 1360COLOUR 129

 1370PRINT TAB(XM,YM);CHR$(228)

 1380COLOUR 3

 1390PRINT TAB(XM+1,YM+1);"AARGH"

 1400FOR I=50 TO 10 STEP -5

 1410SOUND 1,-15,I,2

 1420NEXT I

 1430PRINT TAB(I,25);"IN THE QUICKSAND"

 1440IF RND(1)>.5 THEN PRINT;"A friendly native pulled you out":FOR Q=1 TO 1000:NEXT Q:RETURN

 1450GOTO 2400

 1460REM calculate island

 1470L=RND(3)+10

 1480T=RND(3)+2

 1490W=10+RND(3)

 1500B=RND(2)+17

 1510FOR Y=T TO B

 1520L(Y)=L

 1530R(Y)=L+W

 1540L=L-(SGN (10-Y)*RND(2))

 1550W=W+(SGN (10-Y)*RND(2))

 1560IF L(Y)<1 THEN L(Y)=1

 1570IF R(Y)>30 THEN R(Y)=30

 1580NEXT Y

 1590REM  path logic

 1600X(T+1)=L(T+1)+RND(4)

 1610K=T+2

 1620FOR  P=K TO B-1-RND(3)

 1630X(P)=X(P-1)+RND(3)-2

 1640IF X(P)>=R(P) THEN X(P)=R(P)-1

 1650IF X(P)<=L(P) THEN X(P)=L(P)+1

 1660NEXT P

 1670P=P-1

 1680REM treasure

 1690XT=X(P)

 1700YT=P

 1710REM quicksand logic

 1720FOR Q=1 TO 10

 1730D=RND((P-T-2))+T+1

 1740U(Q)=D

 1750V(Q)=X(D)+SGN (RND(1)-.5)

 1760IF V(Q)<=L(D) THEN V(Q)=V(Q)+3

 1770IF V(Q)>=R(D) THEN V(Q)=V(Q)-3

 1780NEXT Q

 1790RETURN

 1800REM pirate ship

 1810COLOUR 128

 1820COLOUR 3

 1830CLS

 1840R=RND(3)

 1850XS=XS+R

 1860YS=YS+R

 1870PRINT TAB(18,18);CHR$(228)

 1880PRINT TAB(XS,YS);CHR$(226)

 1890PRINT TAB(XS,YS+1);CHR$(227)

 1900FOR Q=1 TO 1000:NEXT Q

 1910IF YS<18 THEN RETURN

 1920PRINT TAB(0,25); "THE PIRATES HAVE LANDED"

 1930PRINT "YOU ARE CAPTURED"

 1940GOTO 2400

 1950REM reprint island

 1960COLOUR 128

 1970CLS

 1980COLOUR 2

 1990FOR X=L(T) TO R(T)

 2000PRINT TAB(X,T-1);CHR$(230)

 2010NEXT X

 2020FOR Y=T TO B

 2030PRINT TAB(L(Y)-1,Y);CHR$(230)

 2040COLOUR 1

 2050FOR X=L(Y) TO R(Y)

 2060PRINT TAB(X,Y);CHR$(230)

 2070NEXT X

 2080COLOUR 2

 2090PRINT TAB(X,Y);CHR$(230)

 2100NEXT Y

 2110FOR X=L(Y-1) TO R(Y-1)

 2120PRINT TAB(X,Y);CHR$(230)

 2130NEXT X

 2140RETURN

 2150GOSUB 810

 2160GOSUB 1250

 2170FOR Q=1 TO 500+RND(500):NEXT Q

 2180GOSUB 1800

 2190GOSUB 1950

 2200GOSUB 1160

 2210RETURN

 2220REM treasure found

 2230DATA 53,61,69,73,81,89,99,101,109,117,121,129,137

 2240DIM N(13)

 2250FOR Q=1 TO 13

 2260READ N(Q)

 2270NEXT Q

 2280MODE 2

 2290VDU 19,7,3,0,0,0

 2300VDU 19,3,7,0,0,0

 2310FOR C=128 TO 135

 2320COLOUR C

 2330CLS

 2340FOR Q=1 TO 200:NEXT Q

 2350SOUND 1,-15,N(C-127),C-124

 2360NEXT C

 2370COLOUR 0

 2380PRINT TAB(0,10);"YOU FOUND THE GOLD"

 2390FOR Q=1 TO 500:NEXT Q

 2400INPUT "Another game (Y/N)",A$

 2410IF LEFT$(A$,1)="Y" THEN RUN

 2420VDU 20

 2430MODE 6

 2440*FX 4,0

Bobsleigh

In this game you have to steer your blue bobsleigh down a random course that winds its way past fir trees. You can choose whether to try a course that is easy to manoeuvre or one that is difficult. (There are actually five levels of difficulty which govern the width of the course.) If you crash you'll hear a dismal tone and that round of the game is over. Play this game to see how adept you are at keeping on course.

How to play

The bobsleigh starts off at the top of the course and the course automatically moves past it. You have to steer the bobsleigh using the right and left arrow keys to ensure that you do not crash into the edges of the course. At the beginning of the game you have to select the difficulty level for the game. This governs the width of the course with 1 producing the widest and therefore easiest, and 5 the narrowest and most difficult.

Subroutine structure


20 
Start of game


30
Defines graphics characters, sets up keyboard and colours


150
Prints first part of track


260
Main play loop


380
Scrolls last part of track off screen


440
Win/lose messages


480
End of game


560
Title frame


670
Moves bobsleigh


770
Prints fir trees

Programming details

The impression of the bobsleigh moving down the course is actually achieved by the course scrolling up the screen past the bobsleigh which only moves to left and right and is at a fixed vertical position. In line 730 the POINT function is used to test whether or not the bobsleigh has hit the side wall. This is done simply by testing what colour is present at the next position that the bobsleigh will be printed at. If the colour is not white then you've crashed into the wall.

The *FX commands in lines 90 and 100 increase the rate at which the keyboard auto-repeats. Normal operation is restored in the end of game routine at line 480. If, for any reason you break out of this program prematurely you will find that the keyboard will still be repeating every key press. To overcome this, you will have to press BREAK and type OLD.

It is also worth noting the use of *FX 15,1 in lines 500 and 690, which has the effect of clearing the Electron's type ahead buffer.

Scope for improvement
If you find this game too fast-moving you can slow it down by increasing the final value of the FOR loops in lines 360 and 420. You might like to add a tune-playing routine to this program like the one given in 'Electron Epsom'.

Program

   10REM Bobsleigh

   20PROCTITLE

   30MODE 1

   40*FX 4,1

   50VDU 19,1,0,0,0,0

   60VDU 19,0,7,0,0,0

   70VDU 19,3,4,0,0,0

   80VDU 19,2,2,0,0,0

   90*FX 12,1

  100*FX 11,1

  110VDU 23,1;0;0;0;0

  120VDU 23,224,&20,&28,&E8,&FC,&7E,&3E,&1E,&0E

  130VDU 23,225,&0F,&0F,&0F,&0F,&0F,&0F,&0F,&0F

  140VDU 23,226,&18,&18,&3C,&3C,&7E,&7E,&18,&18

  150YB=0

  160X=RND(10)+5

  170XB=X+INT(D/2)

  180FOR Y=1 TO 30

  190PRINT TAB(X,31);CHR$(225);TAB(X+D,31);CHR$(225)

  200PROCTREE(X)

  210X=X+SGN(RND(1)-.5)

  220IF X>31 THEN X=31

  230IF X<1 THEN X=1

  240NEXT Y

  250COLOUR 3:PRINT TAB(XB,YB);CHR$(224)

  260FOR Z=1 TO 500:NEXT

  270F=0

  280FOR Z=1 TO 300

  290X=X+SGN(RND(1)-.5)

  300IF X>29 THEN X=29

  310IF X<1 THEN X=1

  320PROCBOB

  330PROCTREE(X)

  340PRINTTAB(X,31);CHR$(225);TAB(X+D,31);CHR$(225)

  350IF F=1 THEN GOTO 460

  360FOR Q=1 TO 10:NEXT Q

  370NEXT

  380FOR Z=1 TO 30

  390PROCBOB

  400PRINT TAB(1,31)'

  410IF F=1 THEN GOTO 460

  420FOR Q=1 TO 10:NEXT Q

  430NEXT Z

  440PRINT TAB(1,20);"CONGRATULATIONS YOU MADE IT"

  450GOTO 480

  460PRINT TAB(1,20);"YOU CRASHED"

  470SOUND 0,-15,2,20

  480*FX 12,0

  490*FX 4,0

  500*FX 15,1

  510INPUT "ANOTHER GAME (Y/N)",A$

  520A$=LEFT$(A$,1)

  530IF A$="Y" THEN RUN

  540CLS

  550END

  560DEF PROCTITLE

  570CLS

  580PRINT TAB(8,1);" B O B S L E I G H"

  590PRINT TAB(4,4);" You must steer your bobsleigh"

  600PRINT TAB(4,7);" down a dangerous course using"

  610PRINT TAB(4,10);" the left and right arrow keys."

  620PRINT TAB(4,14);" Select the difficulty level -"

  630INPUT ''"     from 1 (easy) to 5 (difficult) ",D

  640IF D<1 OR D>5 THEN GOTO 630

  650D=11-D

  660ENDPROC

  670DEF PROCBOB

  680A=INKEY(0)

  690*FX 15,1

  700COLOUR 3

  710IF A=&88 THEN XB=XB-1

  720IF A=&89 THEN XB=XB+1

  730IF POINT(XB*32+20,1024-32*YB-16)=1 OR POINT(XB*32+20,1024-32*YB-48)=1 THEN F=1

  740PRINT TAB(XB,YB);CHR$(224)

  750COLOUR 1

  760ENDPROC

  770DEF PROCTREE(X)

  780IF RND(1)<.4 THEN GOTO 860

  790K=SGN(RND(1)-.5)

  800XT=X+K

  810XT=X+D/2+(K*(RND(5)+D))

  820IF XT<0 OR XT>39 THEN GOTO 780

  830COLOUR 2

  840PRINT TAB(XT,31);CHR$(226);

  850COLOUR 1

  860ENDPROC

7
Capture the Quark

What on earth is a 'quark'? Well may you ask that question, but to find out you'll have to play this game. Here are just a few clues. The game is played on an eight-by-eight chequered board and the object of the game is to trap the quark and prevent him from reaching the bottom of the board. To do this you have two, three or four (determined at random) pieces, or 'quatlins' which can be moved diagonally one square at a time and only up the board. The quark also moves diagonally but he can move both forwards and backwards.

How to play

At the beginning of the game, your quatlines (two, there or four of them according to the luck of the draw) are ranged along the bottom line and the quark is at the top of the board. It is your move first. You'll notice that one of your quatlines is displayed in different colours from the rest. This is the piece that is ready to move. If you want to move another piece, hit any key and control will pass to the next piece in an anti-clockwise direction. Try pressing keys to see this in operation. When you are ready to move a piece, press the left arrow key if you want to move diagonally forward left, and the right arrow key if you want to move diagonally forward right. Once you have moved the quark will make his move automatically and it's your turn again. The Electron will display a message when the game is won, either by you or the quark, but if you want to resign before this, type "R". Just in case you hit this key by mistake you will be given the chance to reconsider and will have to answer "Y" or "N" to the question "Resign?".

Typing tips

The IF statement in line 140 looks as though it's wrong as there are no relational signs. It is, however, correct - the values stored in the array are either '1' or '0' and the Electron treats these as equivalent to 'true' or 'false'.

Subroutine structure


20
Initialises variable


60
Sets up game


90
Main play loop


120
Quark move logic


500
Moves quatlines and validates their moves


650
Selects which quatlin to move


830
Prints differently coloured quatlin


930
Draws board


1120
Draws initial positions of quatlins and quark and initialises board


1390
Defines graphics characters


1590
End of game

Programming details

Although this program runs in a four-colour mode, a hatched black and white character is used to give an extra tone (grey). This is defined at the beginning of subroutine 930. Notice the way that subroutine 830 uses COLOUR to paint a quatlin in two different colours from those normally used for the quatlins.

Program

   10REM Capture the Quark

   20MODE 1

   30DIM D(10,10)

   40DIM X(4),Y(4)

   50DIM A(4),B(4)

   60PROCINIT

   70PROCBOARD

   80H=RND(2)+2:PROCSTART

   90PROCMOVE

  100PROCQUARK

  110GOTO 90

  120DEF PROCQUARK

  130PRINT TAB(0,21)"Quarks Move"

  140IF D(QI+2,QJ+2) AND D(QI+2,QJ) AND D(QI,QJ+2) AND D(QI,QJ) THEN GOTO 1590

  150M=1

  160GOSUB 400

  170IF QI+N<1 OR QI+N>8 THEN GOTO 200

  180IF D(QI+N+1,QJ+M+1) THEN GOSUB 330

  190IF D(QI+N+2,QJ+M+2)=1 AND D(QI+N,QJ+M+2)=1 AND QJ<7 AND QJ>1 THEN M=-M

  200IF D(QI+N+1,QJ+M+1) THEN GOSUB 330

  210IF OI=QI AND OJ=QJ THEN M=-M:N=SGN (RND(1)-.5): GOTO 170

  220OI=QI: OJ=QJ

  230PRINT TAB(QX,QY);SPC(2);TAB(QX,QY+1);SPC(2);

  240QX=QX+N*2

  250QY=QY+M*2

  260PRINT TAB(QX,QY);CHR$(225);CHR$(226);TAB(QX,QY+1);CHR$(227);CHR$(228)

  270D(QI+N+1,QJ+M+1)=2

  280D(QI+1,QJ+1)=0

  290QI=QI+N

  300QJ=QJ+M

  310IF QJ=8 THEN GOTO 1610

  320ENDPROC

  330N=-N

  340IF D(QI+N+1,QJ+M+1)<>1 THEN RETURN

  350M=-M

  360IF QJ<4 THEN N=1

  370IF D(QI+N+1,QJ+M+1)<>1 THEN RETURN

  380N=-N

  390RETURN

  400N=(QI<5)-(QI>=5)

  410R=RND(1)

  420IF QJ>6 THEN RETURN

  430IF R>.5 AND QI>3 THEN GOTO 470

  440IF QI>7 THEN GOTO 460

  450IF (D(QI+3,QJ+3)=0 OR D(QI+2,QJ+3)=0) AND D(QI+2,QJ+2)=0 THEN N=-1: RETURN

  460IF QI<4 OR R>.5 THEN RETURN

  470IF (D(QI-3,QJ+3)=0 OR D(QI-2,QJ+3)=0) AND D(QI-2,QJ+2)=0 THEN N=1: RETURN

  480IF R>.5 THEN GOTO 440

  490RETURN

  500A(HM)=A(HM)+M

  510B(HM)=B(HM)-1

  520IF D(A(HM)+1,B(HM)+1)<>0 THEN GOTO 560

  530D(A(HM)-M+1,B(HM)+2)=0

  540D(A(HM)+1,B(HM)+1)=1

  550GOTO 600

  560A(HM)=A(HM)-M

  570B(HM)=B(HM)+1

  580SOUND 1,-15,80,5

  590GOTO 660

  600PRINT TAB(X(HM),Y(HM));SPC(2);TAB(X(HM),Y(HM)+1);SPC(2);

  610Y(HM)=Y(HM)-2

  620X(HM)=X(HM)+M*2

  630GOSUB 830

  640ENDPROC

  650DEF PROCMOVE

  660PRINT TAB(0,21);"Your Move   "

  670SOUND 1,-15,100,4

  680M$=INKEY$(0)

  690IF M$="" THEN GOTO 680

  700IF M$="R" THEN GOTO 750

  710IF ASC(M$)=&88 THEN M=-1: GOTO 500

  720IF ASC(M$)=&89 THEN M=+1: GOTO 500

  730GOSUB 780

  740GOTO 660

  750INPUT "Resign Y/N",A$

  760IF A$="Y" THEN GOTO 1610

  770GOTO 660

  780GOSUB 890

  790HM=HM+1

  800IF HM>H THEN HM=1

  810GOSUB 830

  820RETURN

  830COLOUR 2

  840COLOUR 128+3

  850PRINT TAB(X(HM),Y(HM));CHR$(229);CHR$(231);TAB(X(HM),Y(HM)+1);CHR$(230);CHR$(232);

  860COLOUR 1

  870COLOUR 128

  880RETURN

  890COLOUR 1

  900COLOUR 128

  910PRINT TAB(X(HM),Y(HM));CHR$(229);CHR$(231);TAB(X(HM),Y(HM)+1);CHR$(230);CHR$(232);

  920RETURN

  930DEFPROCBOARD

  940PRINT TAB(8,3);

  950FOR I=1 TO 4

  960FOR J=1 TO 8

  970PRINT  SPC(2);CHR$(224);CHR$(224);

  980IF J/4=INT (J/4) THEN PRINT:PRINT TAB(8);

  990NEXT J

 1000FOR J=1 TO 8

 1010PRINT CHR$(224);CHR$(224);SPC(2);

 1020IF J/4=INT (J/4) THEN PRINT:PRINT TAB(8);

 1030NEXT J

 1040NEXT I

 1050GCOL 0,1

 1060MOVE 255,415

 1070PLOT 2,515,0

 1080PLOT 2,0,515

 1090PLOT 2,-515,0

 1100PLOT 2,0,-515

 1110ENDPROC

 1120DEF PROCSTART

 1130FOR I=1 TO H

 1140X=I*4+6

 1150PRINT TAB(X,17);CHR$(229);CHR$(231);TAB(X,18);CHR$(230);CHR$(232)

 1160D(I*2+1,9)=1

 1170X(I)=X

 1180Y(I)=17

 1190HM=1

 1200A(I)=I*2

 1210B(I)=8

 1220NEXT I

 1230GOSUB 830

 1240QI=5

 1250QJ=1

 1260QX=QI*2+6

 1270QY=3

 1280PRINT TAB(QX,QY);CHR$(225);CHR$(226);TAB(QX,QY+1);CHR$(227);CHR$
(228)

 1290FOR I=1 TO 10

 1300D(I,1)=1

 1310D(1,I)=1

 1320D(10,I)=1

 1330D(I,10)=1

 1340NEXT I

 1350D(QI+1,QJ+1)=2

 1360OI=0

 1370OJ=0

 1380ENDPROC

 1390DEF PROCINIT

 1400VDU 23,224,&33,&CC,&33,&CC,&33,&CC,&33,&CC

 1410VDU 23,225,&00,&00,&07,&1F,&3F,&73,&73,&7F

 1420VDU 23,226,&00,&00,&E0,&F8,&FC,&CE,&CE,&FE

 1430VDU 23,227,&7F,&3F,&1F,&0F,&07,&03,&01,&00

 1440VDU 23,228,&FE,&FC,&F8,&F0,&E0,&C0,&80,&00

 1450VDU 23,229,&00,&00,&0F,&1F,&1F,&3F,&3B,&73

 1460VDU 23,230,&73,&73,&73,&73,&03,&03,&03,&00

 1470VDU 23,231,&00,&00,&F0,&F8,&F8,&FC,&DC,&CE

 1480VDU 23,232,&CE,&CE,&CE,&CE,&C0,&C0,&C0,&00

 1490*FX 4,1

 1500VDU 23,1;0;0;0;0

 1510VDU 19,0,7,0,0,0

 1520VDU 19,1,0,0,0,0

 1530VDU 19,2,3,0,0,0

 1540VDU 19,3,4,0,0,0

 1550COLOUR 1

 1560COLOUR 128

 1570CLS

 1580ENDPROC

 1590PRINT TAB(0,21)"You Win    "

 1600GOTO 1620

 1610PRINT TAB(0,21)"Quark Wins "

 1620INPUT "Another Game",A$

 1630IF A$="Y" THEN RUN

 1640CLS

 1650PRINT "Bye"

 1660*FX 4,0

8

Commando Jump

This game is a real test of your reaction time and dexterity and is quite compulsive to play. A bright red wall of varying height appears with a little man figure beside it. A countdown "Ready, Steady, GO" is flashed up on the left of the screen and on the word go the man has to jump as high as possible and then scramble up the remainder of the wall. Your success in this game depends entirely on your quick wits and nimble fingers.

How to play

On the word "GO", and no sooner, press any key to make the man jump. The height of the initial jump depends entirely on the delay between the signal appearing and your key press. The quicker you react, the higher than man will jump. The time left to scale the wall is displayed on the screen and while the rest of your five seconds tick away you must keep on pressing any key to get the man over the wall. The man will climb one brick higher for every four key presses - so the more rapidly you press, the more quickly he will climb. If you keep your finger on a key the way will stay where he is - this is because only complete key presses, i.e. press and release, count. If the man is not over within the time limit he will slither back down the wall and you have another try. In all you are given ten attempts. Even if you are very slow off the mark, do press a key - until you do so you cannot move on to the next try. If you hit a key just before the "GO" signal, the computer will accuse you of cheating and you will lose that turn.

Subroutine structure


20
Sets up mode and calls initialisation routine


40
Play loop


60
Countdown


250
Cheat routine


320
Prints wall


420
Jump logic


560
Scramble over remainder of wall routine


680
Man falls back down wall


780
Prints man over wall


900
Zeroes time


920
Sets up screen and defines graphics characters


1000
End of game and messages

Programming details

This is a fairly straightforward application of low resolution dynamic graphics. It runs in Mode 2, a sixteen-colour mode and takes advantage of the Electron's flashing colours. The words "Ready" and "Steady" are made to flash on and off by making them alternate between red and cyan - which causes them to disappear since cyan is used as the background colour. In this game, the Electron's internal clock is used as a reaction timer and a countdown device. The timer counts in one-hundredth of a second intervals and in this game the variable T is scaled to count in seconds. Another interesting point to note is that the repeat failure of the keyboard is disabled by *FX 11,0 in line 980. The repeat key feature is restored at the end of the game by *FX 12,0 in line 1130.

Scope for improvement
You can make this game easier or more difficult by altering the number of key presses needed to scale one brick higher set in line 620. The value is currently 0.25 which means that it takes four keypresses for the man to scale one interval. Decreasing the value will make the game harder and increasing it will make it easier.

Program

   10REM Commando Jump

   20MODE 2

   30PROCINIT

   40PROCWALL

   50PROCFIN

   60COLOUR 9

   70PRINT TAB(0,20);SPC(5)

   80PRINT TAB(0,20);"Ready";

   90PRINT TAB(0,21);SPC(3)

  100FOR I=1 TO RND(1000)+1000

  110NEXT I

  120PRINT TAB(0,20);"Steady"

  130COLOUR 1

  140*FX 15,1

  150FOR I=1 TO RND(1000)+1000

  160NEXT I

  170PRINT TAB(0,20);SPC(6)

  180IF INKEY$(0)<>"" THEN GOTO 250

  190TIME=0

  200SOUND 1,-15,100,5

  210PRINT "GO"

  220IF INKEY$(0)="" THEN GOTO 220

  230T=TIME/100

  240RETURN

  250PRINT TAB(2,10);"Cheat"

  260SOUND 1,-15,3,30

  270FOR K=1 TO 2000:NEXT K

  280PRINT TAB(2,10);SPC(5)

  290TIME=5*100

  300T=TIME/100

  310RETURN

  320DEF PROCWALL

  330COLOUR 6+128

  340CLS

  350JUMP=1

  360H=10+RND(5)

  370FOR I=18 TO 19-H STEP -1

  380COLOUR 1

  390COLOUR 3+128

  400PRINT TAB(10,I);CHR$(224);

  410NEXT I

  420PRINT TAB(10,18-H);CHR$(226);

  430COLOUR 0

  440COLOUR 6+128

  450PRINT TAB(10,20);"Jump ";JUMP

  460PRINT TAB(13,18);CHR$(225);

  470GOSUB 60

  480COLOUR 0

  490IF T>=5 THEN GOTO 740

  500FOR I=18 TO 18-H+INT (T*28) STEP -1

  510PRINT TAB(13,I);" ";

  520PRINT TAB(13,I-1);CHR$(225);

  530SOUND 1,-15,150-I*8,1

  540FOR K=1 TO 20:NEXT K

  550NEXT I

  560J=I: L=INT I

  570T=TIME/100

  580IF T>5 THEN GOTO 680

  590PRINT TAB(5,25);"Time left ";INT ((5-T)*10)/10;"  "

  600IF INKEY$(0)="" THEN GOTO 570

  610PRINT TAB(13,INT L);" ";

  620J=J-0.25

  630L=INT J

  640PRINT TAB(13,L);CHR$(225);

  650IF L<=17-H THEN PRINT TAB(13,L+1);" ":GOTO 780

  660*FX 15,1

  670GOTO 570

  680FOR I=L TO 18

  690PRINT TAB(13,I-1);" "

  700PRINT TAB(13,I);CHR$(225);

  710SOUND 1,-10,200-I,1

  720FOR K=1 TO 10:NEXT K

  730NEXT I

  740JUMP=JUMP+1

  750PRINT TAB(5,25);SPC(15);

  760IF JUMP<=10 THEN GOTO 450

  770ENDPROC

  780FOR I=13 TO 5 STEP -1

  790PRINT TAB(I+1,L);" ";

  800PRINT TAB(I,L);CHR$(225);

  810FOR K=1 TO 100: NEXT K

  820NEXT I

  830FOR I=L TO 18

  840PRINT TAB(5,I-1);" ";

  850PRINT TAB(5,I);CHR$(225);

  860SOUND 1,-15,200-I,1

  870FOR K=1 TO 100:NEXT K

  880NEXT I

  890ENDPROC

  900TIME=0

  910RETURN

  920DEF PROCINIT

  930VDU 23,224,0,&FF,&FF,&FF,&FF,&FF,&FF,&FF

  940VDU 23,225,&18,&18,&FF,&3C,&3C,&24,&24,&24

  950VDU 23,226,&FF,&FF,&FF,&FF,&FF,&FF,&FF,&FF

  960TIME=0

  970VDU 23,1;0;0;0;0

  980*FX 11,0

  990ENDPROC

 1000DEF PROCFIN

 1010IF JUMP<=10 THEN GOTO 1050

 1020PRINT TAB(0,26);"You FAILED !!!"

 1030*FX 15,1

 1040GOTO 1090

 1050*FX 15,1

 1060PRINT TAB(0,27);"You took ";JUMP;

 1070IF JUMP>1 THEN PRINT " jumps" ELSE PRINT" jump"

 1080PRINT "to clear the wall"

 1090PRINT

 1100INPUT "Another game ",A$

 1110IF LEFT$(A$,1)="Y" THEN RUN

 1120CLS

 1130*FX 12,0

9
Electron Epsom

This is a very simple betting game with an impressive and convincing horse race display plus an appropriate musical accompaniment. The tune is 'Camptown Races' and if you've ever heard it before you're sure to recognise it. If you want to show off the graphics and sound capabilities of your Electron this program provides a good demonstration.

How to play

At the beginning of the game you are allocated a hundred chips. You have to bet on which horse will come in first and you must decide how much to stake. The odds are five to one, so if you win having placed 20 you will receive 100. The Electron keeps a tally of your winnings and losses and will tell you if you go broke. During the race, the horse you have backed stands out from the rest of the field - it is the white one. The race is run automatically so there's nothing you can do to make your horse win except cheer it along.
Typing tips

Pay very careful attention to the details of the DATA statements in the function FNTUNE. If you make mistakes in typing in this function, the program may still run but the tune may be unrecognisable or sound discordant! The character between quotes in line 210 is the hash symbol (SHIFT and 3).

Subroutine structure


20
Sets up arrays and defines graphics character


70
Displays title frame with music playing


110
Sets up second mode


200
Prints course and horses


320
Betting routine


430
Main play loop


530
Plays complete tune


640
Plays one note of tune


820
End of game

Programming details

This is the only program in this collection that plays a tune so it is worth drawing your attention to the details of lines 640-730. Think of these DATA statements as being made up of a pair of values. The first in each pair relates to the pitch of the note and the second to the length of time it is sounded for. This second value is typically 40 or 20 or 10 - representing a minim, a crochet or a semiquaver respectively. The number -999 crops up instead of a pitch value every so often. The effect of this is to cause a rest, or pause in the music. Line 730 signals the end of the tune. After detecting 999, 999 the program resets the DATA statements so that next time round the tune starts playing from the beginning. At the beginning of the game the tune is produced by calling subroutine 530 which plays all the notes one after the other. While the race is being run, the notes of the tune are produced by a call directly to subroutine 640 which plays one note in the tune and then moves the pointer to the next note so that the next time it is called the next note is played. To synchronise the sound and the movement, the horses are moved, then subroutine 640 is called to play a note, the horses are moved again, another note is played and so on. The result is a sequence of notes with longer than normal rests between them but, because it does not take much time to move the horses, you still get the impression of a tune being played. You can use this technique in other games but, if the amount of calculation that you have to do between calling each note becomes too long, you'll no longer be able to hear the tune.
Notice that this game uses two display modes. The title frame is in Mode 5 which gives it a bold look. This mode is, however, not suitable for the race itself which is run in Mode 1.

Scope for improvement
If you get tired of the background music you can substitute any tune you like. You could use the same graphics for a Horse Race program in which the player controlled one hose and tried to beat the rest of the field.

Program

   10REM Electron Epsom

   20MODE 5

   30DIM X(5)

   40DIM Y(5)

   50VDU 23,224,&00,&0C,&1A,&FF,&7D,&42,&81,&00

   60VDU 23,226,&0F,&0F,&0F,&0F,&F0,&F0,&F0,&F0

   70CLS

   80PRINT TAB(3,3);"ELECTRON EPSOM"

   90VDU 23,1;0;0;0;0

  100GOSUB 530

  110MODE 1

  120VDU 19,0,2,0,0,0

  130VDU 19,1,0,0,0,0

  140VDU 19,2,3,0,0,0

  150VDU 19,3,7,0,0,0

  160TTAL=100

  170COLOUR 128

  180COLOUR 1

  190CLS

  200FOR X=1 TO 31

  210PRINT TAB(X,1);"*"; TAB(X,11);"*"

  220NEXT

  230FOR Y=2 TO 10

  240PRINT TAB(31,Y);CHR$(226)

  250NEXT

  260COLOUR 2

  270FOR Y=1 TO 5

  280X(Y)=2

  290Y(Y)=Y*2

  300PRINT TAB(X(Y)-1,Y(Y));Y;CHR$(224)

  310NEXT

  320PRINT TAB(0,14);

  330INPUT "WHICH HORSE DO YOU WANT TO BET ON ",B

  340IF B<1 OR B>5 THEN PRINT TAB(0,20);"NO SUCH HORSE":GOTO 320

  350PRINT TAB(0,16);

  360INPUT "HOW MUCH DO YOU WANT TO BET",M

  370IF TTAL-M<0 THEN PRINT TAB(0,20);"YOU DON'T HAVE ENOUGH MONEY!":SOUND 0,-15,50,20:GOTO 350

  380TTAL=TTAL-M

  390PRINT TAB(0,20);STRING$(80," ")

  400VDU 23,1;0;0;0;0

  410COLOUR 3

  420PRINT TAB(X(B),Y(B));CHR$(224)

  430COLOUR 2

  440TEMPO=5

  450Z=RND(5)

  460PRINT TAB(X(Z),Y(Z));" "

  470X(Z)=X(Z)+RND(2)

  480IF Z=B THEN COLOUR 3 ELSE COLOUR 2

  490PRINT TAB(X(Z),Y(Z));CHR$(224)

  500IF X(Z)>30 THEN GOTO 820

  510IF FNTUNE(P,T,TEMPO)=999 THEN RESTORE

  520GOTO 440

  530TEMPO=6

  540I=1

  550IF FNTUNE(P,T,TEMPO)=999 THEN GOTO 610

  560PRINT TAB(I,20);" ";CHR$(224)

  570FOR Z=1 TO 10

  580NEXT

  590I=I+.2

  600GOTO 550

  610PRINT TAB(I+1,15);" "

  620RESTORE

  630RETURN

  640DEF FNTUNE(P,T,TEMPO)

  650DATA 89,20,89,20,89,20,77,20,89,20,97,20,89,20,77,20,-999,20,
77,20,69,60,77,20,69,40

  660DATA 89,20,89,20,77,20,89,20,97,20,89,20,77,20

  670DATA -999,20,77,10,69,10,61,10,69,10,77,20,69,20,61,60

  680DATA -999,20,-999,20,61,30,61,10,77,20,89,20,109,60

  690DATA -999,20,97,30,97,10,109,20,97,20,89,60

  700DATA 77,10,81,10,89,20,89,20,77,10,77,10

  710DATA 89,10,89,10,97,20,89,20,77,40

  720DATA 69,20,77,20,81,10,77,20,69,10,69,10,61,60

  730DATA 999,999

  740READ P,T

  750IF P=999 THEN GOTO 810

  760T=T/TEMPO

  770IF P=-999 THEN P=0:AM=0 ELSE AM=-15

  780SOUND 1,AM,P,T

  790IF ADVAL(-6)<>15 THEN GOTO 790

  800SOUND 1,0,0,1

  810=T

  820IF Z=B THEN PRINT TAB(0,18);" YOU WIN ";INT(5*M):TTAL=TTAL+INT(5*M)

  830IF Z<>B THEN PRINT TAB(0,18);" YOU LOSE ";M

  840IF TTAL<=0 THEN PRINT TAB(0,18);"YOU'RE BROKE   ":FOR Z=1 TO 5000:NEXT:CLS:GOTO 910

  850PRINT "YOU HAVE ";TTAL

  860INPUT "ANOTHER RACE Y/N",A$

  870A$=LEFT$(A$,1)

  880RESTORE

  890CLS

  900IF A$="Y" THEN GOTO 170

  910VDU 20

Guideline

This is a game of skill in which you have to guide a square along a irregular wavy wire. When you play this game with a real wire and a ring, it's a test of how steadily you can guide the ring along the wire. In this Electron version it's a matter of speed as well as hand and eye co-ordination. The square moves from left to right across the screen automatically but you have to keep it on course by pressing the up and down keys. The object of the game is to be on target for as much of the length of the wire as possible and at the end of the game the percentage of time you were successful is displayed.

How to play

At the beginning of the game you have to select the level of difficulty of the game. This determines the size of the square that has to be guided along the wire. Once the square appears, a tone sounds to indicate the start of the game and then the square automatically starts to move right. Press the up and down arrow keys to change its direction to ensure that it stays on the wire. Keeping your finger down on an arrow key will keep the square moving in the same direction. If the square leaves the wire, another tone will sound and will continue to sound at every move until you regain a position on the wire.
Typing tips

For this game the rate at which keys auto-repeat is altered to be extremely rapid. This can cause problems if you either break out of a game while it is playing or if you try to RUN the program before it is working perfectly. One way to overcome the problem of having your keyboard run wild is to define one of the user-definable function keys to restore it to normality. For example, try:


* KEY 1 *FX 12,0|M

Of course, you need to have done this before you run into trouble. The alternative method is to press BREAK and then type "OLD".

Subroutine structure


20
Initialisation routine


140
Call to main play loop


150
End of game


250
Initialises variables and prints title frame


480
Plots wire


590
Draws and moves square and calculates amount on target

Programming details

High resolution graphics are used in this program to give the smooth movement needed to test the players' skill. Subroutine 480 is responsible for plotting the line for the wire and the square is constructed by a collection of PLOT statements at the beginning of subroutine 590. The POINT function is used in this subroutine to test whether the square ring is actually on target around the wire. 

Notice the use of *FX 15,1 in line 670. This call flushes the input buffer to ensure that the computer responds to the current keypress and not to one that was stored earlier. The same call is used in the end game routine to empty the buffer before the player responds to the "Another game" question.

Program

   10REM Guideline

   20GOSUB 250

   30*FX 4,1

   40*FX 11,1

   50*FX 12,1

   60VDU 23,1;0;0;0;0

   70GOSUB 480

   80FOR Q=1 TO 500:NEXT Q

   90PRINT TAB(0,1);"GO"

  100SOUND 1,-15,50,10

  110Y=500:  X=48

  120B=Y:  R=60-DF

  130R2=R/2:  V=10

  140GOSUB 590

  150GCOL 4,4

  160*FX 12,0

  170*FX 4,0

  180*FX 15,1

  190PRINT TAB(0,1);"You were on target  ";INT(HIT/(HIT+MISS)*10000)/100;"% of the time"

  200INPUT "Another game ",A$

  210IF LEFT$(A$,1)="Y" THEN RUN

  220IF LEFT$(A$,1)<>"N" THEN GOTO 200

  230CLS

  240END

  250X=50

  260Y=500

  270D=148

  280P=0

  290MODE 1

  300VDU 19,0,3,0,0,0

  310VDU 19,3,0,0,0,0

  320CLS

  330PRINT TAB(10,2);"G U I D E L I N E"

  340PRINT TAB(4,10);"You must guide a ring along the"

  350PRINT TAB(4);"wavy 'wire' using the up and"

  360PRINT TAB(4);"down arrow keys"

  370PRINT 'TAB(4);"You will be marked on how"

  380PRINT TAB(4);"accurate you are"

  390PRINT TAB(4,25);

  400INPUT "Select the difficulty level"'"    1 (easy) to 5 (difficult) ",DF

  410IF DF<1 OR DF>5 THEN GOTO 390

  420COLOUR 3

  430COLOUR 132

  440DF=DF*4

  450CLS

  460GCOL 4,4

  470RETURN

  480PLOT 4,X,Y

  490FOR I=1 TO 10

  500R=D-RND(2*D)-1

  510Y=Y+R

  520IF Y>800 THEN  Y=Y-R:  R=-R

  530IF Y<50 THEN  Y=Y+R:  R=-R

  540PLOT 9,100,R

  550NEXT I

  560HIT=0

  570MISS=0

  580RETURN

  590PLOT 4,X-R2,Y-R2

  600PLOT 1,0,R

  610PLOT 1,R,0

  620PLOT 1,0,-R

  630PLOT 1,-R,0

  640A$=INKEY$(0)

  650IF ASC(A$)=&8A AND Y-R2>2*V THEN B=B-2*V

  660IF ASC(A$)=&8B AND Y+R<900-2*V THEN B=B+2*V

  670*FX 15,1

  680PLOT 4,X-R2,Y-R2

  690PLOT 1,0,R

  700PLOT 1,R,0

  710PLOT 1,0,-R

  720PLOT 1,-R,0

  730X=X+V

  740IF X>1050 THEN RETURN

  750Y=B

  760F=0

  770FOR I=-R2+4 TO R2-4 STEP 4

  780IF POINT(X,Y+I)=3 THEN  HIT=HIT+1

  790NEXT I

  800IF F=1 THEN GOTO 590

  810MISS=MISS+1

  820SOUND 1,-15,10,5

  830GOTO 590

Magic Dice

Before the days of the micro, family games usually meant one of two things - card games or board games that involved dice. In our family we often couldn't find the dice and we spent ages hunting through drawers and cupboards before we could start our game. Equally often, in the excitement of the game, the dice would end up rolling over the floor and our game would be interrupted as we retrieved it from dark corners.

You may think there's no place for your old games of Ludo and Monopoly now you have a Electron to absorb you, but think again. They are actually very enjoyable games for lots of players especially if you don't have to spend too much time hunting for the dice or, worse still, arguing about which way up it actually fell! Such problems can be solved if you let your Electron join in the game and take over from the dice.

Of course, your Magic Dice can become the centre of a game. You can device gambling games to play against the computer or against other people. After all, dice have been around for thousands of years so there must be plenty of ideas about how to use them.

However you choose to use the program, it will give you a large clear display on the TV screen - colourful, too, if you run it on a colour set. Notice the realistic way the dice actually slows down before it comes to a final halt and the use of sound effects to emphasise this feature.

How to use this program
Using the program is simplicity itself. Type RUN and, when your Electron prompts, just press any key in order to start the dice rolling and then it carries on rolling for a random number of turns and slows down and stops when it is ready. The tone that sounds when the dice has stopped is longer than the one that accompanies each turn. When you've finished with the program, press the BREAK key to stop it running.

Subroutine structure


20
Selects mode and disables cursor


40
Set-up routine


100
Main play loop


260
Prints and unprints dots


430
Draws yellow square for dice


510
Defines graphics characters and sets colours


590
Emits beep sound

Programming details

The essence of a dice program is in generating random numbers. In fact, this program uses random numbers in two ways. Firstly, randomness is used in the conventional way, to determine which face of the dice will show at the next turn - this is done in line 190, which uses the RND function to select 'R', a number between one and six. This information is then used in the printing subroutine (starting at line 260). The program goes to one of the six line numbers 280, 300, 330, 350, 380 or 400, according to the value of 'R', obeying a computed GOTO instruction which has the syntax:

ON (result of arithmetic expression) GOTO

At 280 one dot is printed, at 300 two dots are printed and so on.
The other use of the random number generator is to give the dice realistic suspense. When a human throws a dice it will turn just a few times or quite a number of times, and will slow down before it actually stops. This program copies both these features by incorporating lines 130 and 150-160. Another random number, 'T', with a value between 5 and 12 is selected. This governs the number of turns the dice makes and the pause before the dots reappear lengthens each time it rolls over.

Program

   10REM Magic Dice

   20MODE 5

   30VDU 23,1;0;0;0;0

   40GOSUB 510

   50GOSUB 430

   60COLOUR 128

   70COLOUR 3

   80PRINT TAB(3,28);"press any key"

   90PRINT TAB(3);"to throw"

  100IF INKEY(0)=-1 THEN GOTO 100

  110COLOUR 128+1

  120COLOUR 2

  130T=RND(7)+5

  140FOR I=1 TO T

  150TIME=0

  160REPEAT:UNTIL TIME>2+I*8

  170COLOUR 1

  180GOSUB 260

  190R=RND(6)

  200COLOUR 2

  210GOSUB 260

  220NEXT I

  230*FX 15,1

  240SOUND 1,-15,0,4

  250GOTO 100

  260GOSUB 590

  270ON R GOTO 280,300,330,350,380,400

  280PRINT TAB(9,14);CHR$(224);

  290RETURN

  300PRINT TAB(4,5);CHR$(224);

  310PRINT TAB(14,22);CHR$(224);

  320RETURN

  330GOSUB 280

  340GOTO 300

  350PRINT TAB(4,22);CHR$(224);

  360PRINT TAB(14,5);CHR$(224)

  370GOTO 300

  380GOSUB 350

  390GOTO 280

  400PRINT TAB(4,14);CHR$(224);

  410PRINT TAB(14,14);CHR$(224);

  420GOTO 350

  430COLOUR 1

  440FOR I=1 TO 21

  450PRINT TAB(3,3+I);STRING$(13,CHR$(225))

  460NEXT I

  470R=1

  480COLOUR 128+1

  490COLOUR 2

  500GOTO 260

  510VDU 23,224,&3C,&7E,&FF,&FF,&FF,&FF,&7E,&3C

  520VDU 23,225,&FF,&FF,&FF,&FF,&FF,&FF,&FF,&FF

  530VDU 19,0,2,0,0,0

  540VDU 19,1,3,0,0,0

  550VDU 19,2,1,0,0,0

  560VDU 19,3,0,0,0,0

  570CLS

  580RETURN

  590SOUND 1,-15,0,1

  600RETURN

Positron Invaders
Invaders from the war-like planet Positron are heading towards earth in two types of alien ships and you and your Electron have to defend civilisation as we know it. Your task is daunting - you have to ensure that none of the aliens gets within firing range of earth. The point of no return is marked with an 'X' on the left of the screen. Once any of the advancing ships pass this point the game is over - you will have lost. Your only hope is to wipe out the aliens with your missiles. Every missile that hits its target increases your chance of saving the world.
This game is especially exciting to play because of the use of a two-tone throbbing sound that decreases in pitch as the invaders get closer. If you play this game on a colour TV you'll see that the ships are red and blue.

How to play

You can move your missile launcher left and right using the appropriate arrow keys. Press the up arrow key to fire a missile. You score points for every alien you hit, the ones further away from you counting for more points than the nearer ones. The game is over when you have destroyed all the invaders or when the nearest remaining ones reach the point marked by the X.

Subroutine structure


20
Sets up screen and defines graphics characters


150
Initialises variable


220
Initialises strings


270
Main play loop


580
End of game


740
Moves and fires missile


880
Fire routine


1250
Tests whether alien hit


1360
Moves invaders to left and right

Programming details

The alien ships are stored in strings. The front row (line 220) consists of eight of CHR$(224) with blanks after each of them. The second row (line 230) consists of eight of CHR$(225) arranged, by the simple device of printing a blank before each graphics character, so that the ships are staggered in relation to those in the first row. The third row (line 240) repeats the first, and the back row (line 250) repeats the second. Line 260 sets up a string of blank spaces equivalent to the length of each of the rows of invaders. This is used in lines 520 to 550 to detect whether any of the strings still contain aliens when they reach the critical screen location marked by the 'X'.

The ways the four rows of ships move is interesting as it increases the difficulty of the game. The front row moves to the right, while the second row moves to the left and the back two rows oscillate from side to side. Line 1260 detects when an alien invader is hit. When this happens its position in the string is replaced by a blank space. This manipulation is carried out in line 1270 which divides the string at the appropriate point, inserts a space in place of the destroyed ship and then re-joins the two halves of the string.

One other point to note in this program is the way in which the sound buffer is flushed in line 310 by the presence of a 1 as the third parameter of the first section of the SOUND command. This has the effect of synchronising the noise of the ships approaching closer with the strings moving one position closer. The same technique is used in line 1320 so that a new explosion sound can commence each time an alien is hit.
Scope for improvement
You might like to add a routine to make the aliens shoot back at random so that the missile launcher faces the added problem of dodging enemy fire. If you want to make the game easier by lengthening the time before the aliens ships move forward in a row, increase the value to the right of the > sign in line 310.

Program

   10REM Positron Invaders

   20MODE 5

   30VDU 19,0,3,0,0,0

   40VDU 19,1,4,0,0,0

   50VDU 19,2,1,0,0,0

   60VDU 19,3,2,0,0,0

   70*FX 4,1

   80*FX 12,1

   90*FX 11,1

  100VDU 23,1;0;0;0;0

  110VDU 23,224,&18,&3C,&7E,&FF,&C3,&C3,&66,&24

  120VDU 23,225,&18,&3C,&7E,&FF,&3C,&66,&C3,&66

  130VDU 23,226,&18,&18,&18,&3C,&7E,&7E,&FF,&FF

  140VDU 23,227,&28,&88,&91,&28,&1C,&34,&A4,&A4

  150Y=1

  160XL=10

  170YM=0

  180T=0

  190S=0

  200J=0

  210K=0

  220A$=STRING$(8,CHR$(224)+" ")

  230B$=STRING$(8," "+CHR$(225))

  240C$=A$

  250D$=B$

  260E$=STRING$(16," ")

  270COLOUR 3

  280PRINT TAB(0,14);"X"

  290J=NOT J

  300IF K=1 THEN GOTO 600

  310IF T>40+RND(15) THEN PRINT TAB(1,Y);E$:Y=Y+2:T=0:SOUND &0011,0,0,1

  320IF J THEN A$=FNM(A$) ELSE A$=FNR(A$)

  330COLOUR 2

  340PRINT TAB(1,Y);A$

  350SOUND 1,-15,121-Y*8,5

  360PROCMOVE

  370IF J THEN B$=FNM(B$) ELSE B$=FNR(B$)

  380COLOUR 1

  390PRINT TAB(1,Y+2);B$

  400SOUND 1,-15,129-Y*8,5

  410PROCMOVE

  420C$=FNM(C$)

  430COLOUR 2

  440PRINT TAB(1,Y+4);C$

  450SOUND 1,-15,121-Y*8,5

  460PROCMOVE

  470D$=FNR(D$)

  480COLOUR 1

  490PRINT TAB(1,Y+6);D$

  500SOUND 1,-15,129-Y*8,5

  510PROCMOVE

  520IF Y>8 AND D$<>E$ THEN GOTO 580

  530IF Y>10 AND C$<>E$ THEN GOTO 580

  540IF Y>12 AND C$<>E$ THEN GOTO 580

  550IF Y>14 AND A$<>E$ THEN GOTO 580

  560T=T+1

  570GOTO 270

  580PRINT TAB(1,23);" THEY GOT YOU!!"

  590GOTO 610

  600PRINT TAB(1,23);"WELL DONE !"'"YOU SAVED THE WORLD!"

  610*FX 15,1

  620*FX 4,0

  630*FX 12,0

  640SOUND &0011,0,0,1

  650IF K=0 THEN SOUND &0010,-15,4,20

  660INPUT "ANOTHER GAME Y/N",A$

  670A$=LEFT$(A$,1)

  680IF A$="Y" THEN RUN

  690*FX 4,0

  700*FX 12,0

  710VDU 20

  720MODE 7

  730END

  740DEF PROCMOVE

  750A=INKEY(0)

  760*FX 15,1

  770T=T+1

  780COLOUR 3

  790PRINT TAB(XL,21);CHR$(226)

  800IF A=-1 THEN ENDPROC

  810PRINT TAB(XL,21);" "

  820IF A=&88 AND XL>1 THEN XL=XL-1

  830IF A=&89 AND XL<16 THEN XL=XL+1

  840COLOUR 3

  850PRINT TAB(XL,21);CHR$(226)

  860IF A=&8B THEN PROCFIRE

  870ENDPROC

  880DEF PROCFIRE

  890COLOUR 3

  900FOR M=19 TO Y+6 STEP -1

  910PRINT TAB(XL,M);":";

  920PRINT TAB(XL,M+1);" "

  930NEXT

  940PRINT TAB(XL,M+1);" "

  950F=0

  960Q$=D$

  970R=6

  980PROCHIT

  990D$=Q$

 1000IF F=1 THEN GOTO 1220

 1010COLOUR 3

 1020PRINT TAB(XL,Y+5);".";TAB(XL,Y+5);" ";TAB(XL,Y+4);".";TAB(XL,Y+4);" "

 1030Q$=C$

 1040R=4

 1050PROCHIT

 1060C$=Q$

 1070IF F=1 THEN GOTO 1220

 1080COLOUR 3

 1090PRINT TAB(XL,Y+3);".";TAB(XL,Y+3);" ";TAB(XL,Y+2);".";TAB(XL,Y+2);" "

 1100Q$=B$

 1110R=2

 1120COLOUR 3

 1130PROCHIT

 1140B$=Q$

 1150IF F=1 THEN GOTO 1220

 1160PRINT TAB(XL,Y+1);".";TAB(XL,Y+1);" ";TAB(XL,Y);".";TAB(XL,Y);" "

 1170Q$=A$

 1180R=0

 1190COLOUR 3

 1200PROCHIT

 1210A$=Q$

 1220IF A$=E$ AND B$=E$ AND C$=E$ AND D$=E$ THEN K=1

 1230IF Q$=E$ THEN PRINT TAB(1,Y);E$:Y=Y+2

 1240ENDPROC

 1250DEF PROCHIT

 1260IF MID$(Q$,XL,1)=" " THEN ENDPROC

 1270Q$=MID$(Q$,1,XL-1)+" "+MID$(Q$,XL+1)

 1280F=1

 1290S=S+10-Y

 1300COLOUR 3

 1310PRINT TAB(XL,Y+R);CHR$(227)

 1320SOUND &0010,-15,4,3

 1330PRINT TAB(6,30);"SCORE ";S;"   ";

 1340T=T-RND(3)

 1350ENDPROC

 1360DEF FNM(Q$)=MID$(Q$,2)+LEFT$(Q$,1)

 1370DEF FNR(Q$)=RIGHT$(Q$,1)+LEFT$(Q$,LEN(Q$)-1)

Mirror Tile

Although your Electron opens up lots of new possibilities for games, it's good to know that it can also conjure up old favourites. Mirror Tile is a colourful and versatile version of a game that is conventionally played on a small board with the pieces slotted into one another and into their surrounding frame. This construct is vital to the game, the object of which is to rearrange the pieces to match a given pattern - hence its title 'Mirror Tile'.

If you have never played with a tile puzzle, look at the illustration of the game's display. You'll see a four-by-four square of letters and you'll notice that one position is empty. In other words there are fifteen letters and one hole. The hole allows you to move the letters around the board.

How to play

Imagine for a moment that the board was really made of plastic pieces. You could slide a piece that was either above or below or to either side of the hole into it and the position of the piece you moved would then be empty - that is, it would become the hole. Notice that there are only a limited number of possible moves - two, three or four depending on the position of the hole, and that it is impossible to move on the diagonals. 

These same rules apply to the Electron version of the game. You can move any letter that is directly to the left or right of the hole or just above or below it. To indicate your choice you type in the number (1 to 16) of the square containing the piece you want to move. If you try to make a wrong move the Electron won't let you. Instead it will be helpful and number each square for you, in case your mistake was due to typing in the wrong number for your choice. If you want to see these numbers displayed, type any letter key - in fact any key other than one for a legal move.

When you RUN this program, the first thing it asks you is whether you wish to input your own set of words. If you reply "N" then the square will fill with the letters A to O. If you prefer to select your own starting arrangement you will be asked to type in three four-letter words, and one three-letter word. Of course, this means you can vary the difficulty of the puzzle. If you choose words that repeat some letters the game will actually be easier. For example, typing in ROOF, CATS, RENT and TENT would give a fairly simple game. The most difficult puzzle is one where every letter is different, e.g. HOME, CART, WING, SKY.

Once you've typed in your words, you'll see them being shuffled - the program will already have asked you how many shuffles it should perform and the more it shuffles the more difficult you will find it. Then it's your turn - to sort them out again into the initial arrangement. Your Electron will count your moves and let you know how many you took at the end of the game.

Subroutine structure


20
Defines array


50
Sets up game


140
Main play loop


180
End of game


230
Checks for end of game


320
Sets up default (alphabetic) board


460
Prints frame


750
Shuffle routine


900
Prints number overlay


990
Blanks out previous messages


1090
Move logic


1320
Locates empty space


1410
Prints title and initial questions


1550
Performs move


1610
Asks for input of words


1950
Updates positions


2010
Defines graphics characters and sets up screen

Programming details

This program is complicated both because of its length and also because it involves a lot of logic. However, because of the way in which it uses Acorn BASIC's procedures, its structure is very clear and if you follow it through you should be able to see what happens at every step.

Scope for improvement
Adding to a program that is already as long as this one may seem to be a tall order. However, there is actually scope for improvement. A routine that reminds the player of the target arrangement might be a very useful extra.

Program

   10REM Mirror Tile

   20MODE 1

   30DIM B(4,4)

   40DIM B$(16),W$(16)

   50PROCTITLE

   60PROCLET

   70IF WR=1 THEN PROCWORDS

   80MOV=0

   90ER=0

  100PROCGRAPH

  110PROCFRAME

  120PROCHOLE

  130PROCSHUF

  140PROCMOVE

  150PROCFIN

  160MOV=MOV+1

  170IF FIN<>0 THEN GOTO 140

  180PRINT TAB(0,22);"You did it in ";MOV;" moves"

  190INPUT "Another game ",A$

  200IF LEFT$(A$,1)="Y" THEN RUN

  210CLS

  220END

  230DEF PROCFIN

  240FIN=0: K=0

  250FOR I=1 TO 4

  260FOR J=1 TO 4

  270K=K+1

  280IF B$(B(I,J))<>W$(K) THEN FIN=1

  290NEXT J

  300NEXT I

  310ENDPROC

  320DEF PROCLET

  330K=0

  340FOR I=1 TO 4

  350FOR J=1 TO 4

  360K=K+1

  370B$(K)=CHR$(64+K)

  380B(I,J)=K

  390NEXT J

  400NEXT I

  410B$(16)=" "

  420FOR I=1 TO 16

  430W$(I)=B$(I)

  440NEXT I

  450ENDPROC

  460DEF PROCFRAME

  470COLOUR 128+3

  480COLOUR 2

  490FOR I=0 TO 3

  500FOR J=0 TO 3

  510PRINT TAB(10+J*3,4+I*3);SPC(2);CHR$(224)

  520NEXT J

  530FOR J=0 TO 3

  540PRINT TAB(10+J*3,5+I*3);SPC(1);B$(B(I+1,J+1));CHR$(224)

  550NEXT J

  560FOR J=0 TO 3

  570PRINT TAB(10+J*3,6+I*3);CHR$(226);CHR$(226);CHR$(225)

  580NEXT J

  590NEXT I

  600FOR I=0 TO 11

  610COLOUR 2:COLOUR 128+1

  620PRINT TAB(10+I,3);CHR$(227)

  630COLOUR 1:COLOUR 128+2

  640PRINT TAB(10+I,16);CHR$(227)

  650PRINT TAB(9,4+I);CHR$(228)

  660COLOUR 2:COLOUR 128+1

  670PRINT TAB(22,4+I);CHR$(228)

  680NEXT I

  690PRINT TAB(9,3);CHR$(229)

  700PRINT TAB(22,16);CHR$(230)

  710PRINT TAB(22,3);CHR$(231)

  720PRINT TAB(9,16);CHR$(232)

  730COLOUR 128+3

  740ENDPROC

  750DEF PROCSHUF

  760IS=4: JS=4: IY=0:JY=0

  770FOR D=1 TO S

  780I=IS: J=JS

  790IF RND(1)>.5 THEN GOTO 830

  800I=IS+INT (RND(1)*2)*2-1

  810IF I>4 OR I<1 THEN I=IS:GOTO 830

  820GOTO 850

  830J=JS+INT (RND(1)*2)*2-1

  840IF J>4 OR J<1 THEN J=JS:GOTO 780

  850IF I=IY AND J=JY THEN GOTO 780

  860IY=IS: JY=JS

  870PROCDOMOVE

  880NEXT D

  890ENDPROC

  900DEF PROCOVER

  910K=0

  920FOR I=0 TO 3

  930FOR J=0 TO 3

  940K=K+1

  950PRINT TAB(10+J*3,4+I*3);K

  960NEXT J

  970NEXT I

  980ENDPROC

  990DEF PROCBLANK

 1000FOR L=0 TO 3

 1010FOR P=0 TO 3

 1020PRINT TAB(10+P*3,4+L*3);SPC(2)

 1030NEXT P

 1040NEXT L

 1050IF ER=0 THEN ENDPROC

 1060PRINT TAB(0,20);SPC(64)

 1070ER=0

 1080ENDPROC

 1090DEF PROCMOVE

 1100PRINT TAB(0,22);"What is your move ";

 1110INPUT LINE M$

 1120IF M$="" THEN GOTO 1090

 1130IF LEN(M$)<2 THEN M$="0"+M$

 1140IF MID$(M$,1,1)<"0" OR MID$(M$,1,1)>"9" OR MID$(M$,2,1)<"0" OR MID$(M$,2,1)>"9" THEN PROCOVER:GOTO 1090

 1150M=VAL(M$)

 1160IF M>0 AND M<17 THEN GOTO 1220

 1170ER=1

 1180PRINT TAB(0,20); "A move must be a number between "

 1190PRINT TAB(0,21);"1 and 16 as shown"

 1200PROCOVER

 1210GOTO 1090

 1220I=INT ((M-1)/4)

 1230J=M-I*4

 1240I=I+1

 1250IF ABS (I-IS)+ABS (J-JS)=1 THEN PROCDOMOVE:ENDPROC

 1260SOUND 1,-10,0,3

 1270ER=1

 1280PRINT TAB(0,20);"You can only move a tile next to"

 1290PRINT TAB(0,21); "the space";SPC(10)

 1300PROCOVER

 1310GOTO 1090

 1320DEF PROCHOLE

 1330K=0

 1340FOR L=1 TO 4

 1350FOR P=1 TO 4

 1360K=K+1

 1370IF B(P,L)=16 THEN IS=L:JS=P:MS=K

 1380NEXT P

 1390NEXT L

 1400ENDPROC

 1410DEF PROCTITLE

 1420CLS

 1430PRINT TAB(5);"M i r r o r   T i l e"

 1440PRINT TAB(0,10);"Do you want to input your"

 1450PRINT "own set of words ";

 1460INPUT A$

 1470IF A$="" THEN GOTO 1460

 1480IF LEFT$(A$,1)="Y" THEN WR=1: GOTO 1510

 1490IF LEFT$(A$,1)="N" THEN WR=0: GOTO 1510

 1500GOTO 1460

 1510PRINT TAB(0,15);"How many shuffles ";

 1520INPUT S

 1530IF S<1 THEN GOTO 1510

 1540ENDPROC

 1550DEF PROCDOMOVE

 1560PROCBLANK

 1570PROCUPDATE

 1580PRINT TAB(11+(J-1)*3,5+(I-1)*3);B$(B(I,J))

 1590PRINT TAB(11+(JS-1)*3,5+(IS-1)*3);B$(B(IS,JS))

 1600ENDPROC

 1610DEF PROCWORDS

 1620CLS

 1630PRINT TAB(0,5);"Choose 3 four-letter words"

 1640PRINT "and 1 three-letter word."

 1650INPUT "Type the first four-letter word ",A$

 1660IF LEN(A$)<>4 THEN GOTO 1650

 1670FOR I=1 TO 4

 1680W$(I)=MID$(A$,I,1)

 1690NEXT I

 1700PRINT "First word= ";A$

 1710INPUT "Type the second four-letter word", A$

 1720IF LEN(A$)<>4 THEN GOTO 1710

 1730FOR I=5 TO 8

 1740W$(I)=MID$(A$,I-4,1)

 1750NEXT I

 1760PRINT "Second word= ";A$

 1770INPUT "Type the third four-letter word ",A$

 1780IF LEN(A$)<>4 THEN GOTO 1770

 1790FOR I=9 TO 12

 1800W$(I)=MID$(A$,I-8,1)

 1810NEXT I

 1820PRINT "Third word= ";A$

 1830INPUT "Type the three-letter word ",A$

 1840IF LEN(A$)<>4 THEN GOTO 1830

 1850FOR I=13 TO 15

 1860W$(I)=MID$(A$,I-12,1)

 1870NEXT I

 1880PRINT "Fourth word= ";A$

 1890FOR I=1 TO 1000:NEXT I

 1900W$(16)=" "

 1910FOR I=1 TO 16

 1920B$(I)=W$(I)

 1930NEXT I

 1940ENDPROC

 1950DEF PROCUPDATE

 1960B(IS,JS)=B(I,J)

 1970B(I,J)=16

 1980T=IS: IS=I: I=T

 1990T=J: J=JS: JS=T

 2000ENDPROC

 2010DEF PROCGRAPH

 2020VDU 23,224,1,1,1,1,1,1,1,1

 2030VDU 23,225,1,1,1,1,1,1,1,&FF

 2040VDU 23,226,0,0,0,0,0,0,0,&FF

 2050VDU 23,227,0,0,0,0,&FF,&FF,&FF,&FF

 2060VDU 23,228,&F0,&F0,&F0,&F0,&F0,&F0,&F0,&F0

 2070VDU 23,229,0,0,0,0,&0F,&0F,&0F,&0F

 2080VDU 23,230,&F0,&F0,&F0,&F0,0,0,0,0

 2090VDU 23,231,0,0,0,0,&F0,&F0,&F0,&F0

 2100VDU 23,232,&0F,&0F,&0F,&0F,0,0,0,0

 2110VDU 19,0,7,0,0,0

 2120VDU 19,1,1,0,0,0

 2130VDU 19,2,4,0,0,0

 2140VDU 19,3,3,0,0,0

 2150COLOUR 128+3

 2160COLOUR 2

 2170CLS

 2180ENDPROC

Pot Shot

This game provides the ideal type of target practice. However many times you score a direct hit, the magic bird continues to fly on, allowing you to take aim and fire again and again. The elements of this game are simple - a blue sky with a single white cloud, a bird winging its way from left to right and your rifle sight. The object is straightforward - to line up the sight with the bird and shoot it. But in practice it's not that simple. For one thing, every time you fire, your rifle 'kicks' to one side or the other so you have to realign your sight for another shot. Also, the bird (and your sight too) disappear behind the cloud when they reach it. A total of five birds fly across the sky and there is no limit to the number of hits you can score - your total for each bird and the whole game are displayed at the end. There is a sound every time you fire your rifle and a distinctively different sound when you hit the bird.

How to play

To hit the target you have to line the cross point of your sight up with the centre of the bird. Use all four arrow keys to move your sight and press 'F' to fire. There are five birds in all and you may hit each one as often as you can.

Subroutine structure


20 
Sets mode and call initialisation routine


40
Main play loop


110
Plots cross


180
Moves sight, calls firing routine and moves bird


390
Draws cloud and sets up display


580
Draws bird


650
Shoots, tests for hit and recoils sight


720
Sets up screen, defines graphics characters and envelope


940
End of game

Programming details

The rifle shot sound is produced using the ENVELOPE command defined in line 920. It is used in the SOUND statement in line 660 where it modifies a noise produced on channel 0 (the noise channel). The same envelope is then used in line 690 where it modifies a tone produced on channel 1 to give the sound that indicates a hit. The use of high resolution graphics in this program means that the range and smoothness of both the bird and the rifle sight is better than could be achieved with low resolution graphics. The function POINT is used in line 670 to discover if the target has been hit. Notice the extensive use of GCOL 4,4 to plot and unplot high resolution shapes. It is also interesting to note the way the bird's flight path is calculated and stored in the array B in line 840 for use later in the program.

Scope for improvement
To make this program even more of a challenge you could add more than one cloud and allow the bird to have more than one path across the sky.

Program

   10REM Pot Shot

   20MODE 5

   30PROCINIT

   40FOR G=1 TO 5

   50PROCSTART

   60PRINT TAB(5,25); "Bird ";G;

   70PROCMOVE

   80NEXT G

   90GOTO940

  100END

  110DEF PROCCROSS

  120PLOT 4,X-30,Y

  130PLOT 1,60,0

  140PLOT 4,X,Y-30

  150PLOT 1,0,60

  160PLOT 64+5,X,Y

  170ENDPROC

  180DEF PROCMOVE

  190A$=INKEY$(0)

  200*FX 15,1

  210GCOL 4,4

  220PROCCROSS

  230J=B(I)

  240PROCBIRD

  250IF ASC(A$)=&88 AND X>38 THEN  X=X-8

  260IF ASC(A$)=&8A AND Y>250 THEN  Y=Y-8

  270IF ASC(A$)=&8B AND Y<1000 THEN  Y=Y+8

  280IF ASC(A$)=&89 AND X<1268 THEN  X=X+8

  290I=I+.75

  300J=B(I)

  310GCOL 4,4

  320PROCBIRD

  330IF A$="F" THEN PROCFIRE

  340*FX 15,1

  350PROCCROSS

  360IF FIN=1 THEN FIN=0:ENDPROC

  370FOR Z=1 TO 10:NEXT Z

  380GOTO 190

  390DEF PROCSTART

  400COLOUR 128+1

  410CLS

  420COLOUR 128

  430J=RND(5)

  440R=RND(10)-1

  450PRINT TAB(R+5,J);SPC(3);

  460PRINT TAB(R+4,J+1);SPC(5);

  470COLOUR 128+1

  480FIN=0

  490X=RND(300)+200

  500Y=800

  510I=3

  520COLOUR 2

  530GCOL 4,4

  540PROCCROSS

  550J=B(I)

  560PROCBIRD

  570ENDPROC

  580DEF PROCBIRD

  590VDU 5

  600MOVE I*5,J

  610PRINT CHR$(224);CHR$(225);

  620VDU 4

  630IF I>235 THEN LET FIN=1

  640ENDPROC

  650DEF PROCFIRE

  660SOUND &10,1,5,2

  670IF POINT(X,Y)<>2 THEN X=X+50-RND(100):ENDPROC

  680X=X+50-RND(100)

  690SOUND &11,1,100,3

  700H(G)=H(G)+1

  710ENDPROC

  720DEF PROCINIT

  730HIT=0

  740COLOUR 2

  750VDU 23,1;0;0;0;0

  760CLS

  770PRINT TAB(3,10);"P O T  S H O T"

  780VDU 19,0,7,0,0,0

  790VDU 19,1,6,0,0,0

  800VDU 19,2,0,0,0,0

  810VDU 19,3,7,0,0,0

  820DIM B(254)

  830FOR I=1 TO 250

  840B(I)=1000-(125-I)*(125-I)/40

  850NEXT I

  860DIM H(5)

  870VDU 23,224,&00,&00,&00,&00,&18,&24,&43,&81

  880VDU 23,225,&00,&00,&00,&00,&18,&24,&C2,&81

  890*FX 4,1

  900*FX 11,1

  910*FX 12,1

  920ENVELOPE 1,1,1,0,0,6,0,0,126,0,0,-126,126,126

  930ENDPROC

  940MODE 6

  950T=0

  960FOR I=1 TO 5

  970PRINT TAB(10,I+5);"Bird ";I;" HIT ";H(I)

  980T=T+H(I)

  990NEXT I

 1000*FX 12,0

 1010*FX 15,1

 1020PRINT TAB(10,12);"Total hits= ";T

 1030PRINT TAB(10,14);

 1040INPUT "Another game Y/N ",A$

 1050IF A$="Y" THEN RUN

 1060*FX 4,0

Save The Whale

This is a moving graphics game for conservationists! The object of the game is to ensure that the whale survives to swim on in artic seas. You have to outwit the eskimos who are hunting the whale in their kayaks. If they run into the icebergs they will have to abandon their hunt, so you must lure them towards these obstacles by moving the whale in such a way that, in approaching it, the eskimos crash.

How to play

At the beginning of the game you can select the difficulty level for your turn. Your selection governs the starting positions of the icebergs and so makes the game easier or harder to play. To move the whale you press any of the arrow keys. If any kayak runs into an iceberg, the kayak vanishes. If the whale runs into an iceberg, the iceberg vanishes - this of course reduces his protection so it is not advisable except in extreme circumstances - and if an eskimo reaches the whale, he harpoons the whale and kills him. The game is over when all the eskimos have been removed from play or when the whale is dead.
Subroutine structure


20
Sets up graphics characters and arrays


150
Title frame


250
Sets up screen and turns graphics cursor off


280
Prints kayaks


350
Prints icebergs


410
Prints whale


450
Main play loop


520
Checks for game over


540
Move whale routine


680
Move kayaks routine


790
End of game

Programming details

The initial positions of the kayaks are set at random within a band at the edges of the screen (lines 290 and 300). The initial positions of the icebergs are set in a similar fashion (lines 380 and 380) but account is also taken of the difficulty factor, 'D' input at 230. The POINT function is used in line 740 to detect whether a kayak has landed on an iceberg - in which case that is the end of the kayak. The alternative method of simply comparing co-ordinates is used in line 750 to detect whether a kayak has landed on the whale - in which case that is the end of the whale (and the game).

Program

   10REM Save the Whale

   20MODE 1

   30*FX 4,1

   40DIM X(20)

   50DIM Y(20)

   60DIM U(20)

   70DIM V(20)

   80VDU 23,224,&00,&00,&30,&78,&F9,&FF,&F9,&00

   90VDU 23,225,&00,&20,&72,&76,&7E,&FF,&FF,&FF

  100VDU 23,226,&00,&C8,&58,&38,&FF,&3C,&08,&06

  110VDU 19,0,6,0,0,0

  120VDU 19,1,7,0,0,0

  130VDU 19,2,1,0,0,0

  140VDU 19,3,0,0,0,0

  150CLS

  160PRINT TAB(2,2);"S A V E  T H E  W H A L E"

  170PRINT ''''"In this game, you, the whale    ";CHR$(224)

  180PRINT '"must outwit the eskimos hunting"

  190PRINT '"you in their kayaks    ";CHR$(226)

  200PRINT '"by luring them onto the icebergs   ";CHR$(225)

  210PRINT TAB(0,20);"WHICH DIFFICULTY LEVEL DO YOU WISH TO"

  220PRINT '"PLAY AT"

  230INPUT '"(1) EXPERT,(2) INTERMEDIATE,(3) NOVICE ",D

  240IF D<1 OR D>3 THEN GOTO 230

  250CLS

  260VDU 23,1;0;0;0;0

  270COLOUR 2

  280FOR C=1 TO 20

  290X=SGN(RND(1)-.5)*(RND(4)+10)+15

  300Y=SGN(RND(1)-.5)*(RND(4)+6)+11

  310PRINT TAB(X,Y);CHR$(226)

  320X(C)=X

  330Y(C)=Y

  340NEXT C

  350COLOUR 1

  360FOR C=1 TO 20

  370U(C)=SGN(RND(1)-.5)*(RND(4)+4-D)+15

  380V(C)=SGN(RND(1)-.5)*(RND(4)+3-D)+9

  390PRINT TAB(U(C),V(C));CHR$(225)

  400NEXT C

  410COLOUR 3

  420X=INT(RND(2)+10)

  430Y=INT(RND(2)+10)

  440PRINT TAB(X,Y);CHR$(224)

  450GOSUB 540

  460F=0

  470FOR C=1 TO 20

  480IF X(C)=0 THEN GOTO 510

  490F=1

  500GOSUB 680

  510NEXT C

  520IF F=0 THEN GOTO 830

  530GOTO 450

  540SOUND 0,-15,3,10

  550Z=X:V=Y

  560*FX 15,1

  570A=INKEY(0)

  580IF A=-1 THEN GOTO 570

  590IF A=&88 AND X>1 THEN X=X-1

  600IF A=&89 AND X<31 THEN X=X+1

  610IF A=&8A AND Y<21 THEN Y=Y+1

  620IF A=&8B AND Y>1 THEN Y=Y-1

  630COLOUR 4

  640PRINT TAB(Z,V);" "

  650COLOUR 3

  660PRINT TAB(X,Y);CHR$(224)

  670RETURN

  680PRINT TAB(X(C),Y(C));" "

  690E=0

  700E=SGN(X(C)-X)

  710X(C)=INT(X(C)-E)

  720E=SGN(Y(C)-Y)

  730Y(C)=INT(Y(C)-E)

  740IF POINT(X(C)*32+16,1023-32*Y(C)-18)=1 THEN X(C)=0:SOUND 1,-15,200,1:GOTO 780

  750IF X(C)=X AND Y(C)=Y THEN GOTO 790

  760COLOUR 2

  770PRINT TAB(X(C),Y(C));CHR$(226)

  780RETURN

  790COLOUR 3

  800PRINT TAB(X(C),Y(C));" "

  810PRINT TAB(1,20);" YOU WERE KILLED"

  820GOTO 840

  830PRINT TAB(1,20);"YOU ESCAPED THIS TIME"

  840INPUT "ANOTHER GAME (Y/N)",A$

  850A$=LEFT$(A$,1)

  860IF A$="Y" THEN RUN

  870*FX 4,0

  880MODE 6

  890CLS

  900END

Mighty Missile

Your weapon can destroy anything - anything that it actually hits. So the only problem in this game is to ensure that the missile finds its target, quickly and accurately. The enemy skips sweep in from the left and the right, firing relentlessly. Your missile in only vulnerable if its protective shield is eroded away and then it can be easily blasted in its home base. Otherwise, it is impervious to enemy fire, even outside its base. If it hits an enemy it will explode on contact, but if it fails to find its target it will disintegrate as it reaches the upper atmosphere. You can launch ten missiles and there are ten enemy ships. Each ship maintains a stable orbit until you actually take a shot at it so you can wait in the base while deciding which side to fire from and when to fire - except that with every orbit more of your shield is blasted away by enemy fire and once there is only 20 per cent of it left you will no longer have any protection from the enemies' lasers. This fast moving graphics game is enhanced by sound effects and is quite compulsive to play.

How to play

In this game it is important to notice how much 'shield strength' you have left since when the figure displayed drops to 20 per cent you will be vulnerable to attack. Once the shield is so eroded, the enemy laser is able to home in and destroy you wherever they hit you, including inside the missile base. The object of the game is to score as many hits as possible so it is worth watching each new enemy ship's orbit at least once or twice before you try to shoot it down. To fire you have to leave your base. Do this by pressing the right or left arrow key. This will take you to a fixed position on the right or left of the screen and launch the missile. Remember to take into account the time it will take for your missile to reach the enemy ship which will continue on its path! At the end of the game your score is displayed and you are given the option of another game.

Typing tips

In line 690, notice the space after the percentage sign and before the double quotes. This serves the important function of blanking out previous figures as the number displayed gets smaller and so occupies fewer positions on the screen.

Subroutine structure


20
Set up


70
Main play loop

310
End of game


380
Prints attacker and fires laser


500
Checks to see if player has activated missile


620
Calculates shield strength


710
Moves and fires missile and tests for hit or miss


780
Explosion graphics and sound


940
Sets up attack orbit


1070
Prints shield


1200
Defines graphics characters and sets up screen display

Programming details

As this program has a clear structure and uses separate procedures for most of its major elements, you should find it relatively easy to follow. One interesting point to note is the way in which the  path of the attacking ship is calculated in procedure PROCPATH and stored in a pair of arrays X and Y to be used repeatedly for the various orbits used during the game. A second point of interest is that although all the graphics used are low resolution graphics the laser zap from the attaching spaceship is a high resolution graphics command which will blank out any black points that it passes through. So although the shield is initially printed using low resolution blocks, it is whittled away by the laser beam passing through it. The strength of the shield is estimated, in subroutine 620, by the number of black points left in the shield, using the POINT function. POINT is 1 if the point at the x,y co-ordinate is black and 0 if it is blue.

Program

   10REM Mighty Missile

   20MODE 1

   30PROCINIT

   40PROCPATH

   50PROCBLOCK

   60PROCSTREN

   70FOR A=1 TO 10

   80DIR=SGN (RND-.5)

   90PRINT TAB(21,1);"Attacker ";A

  100PRINT TAB(21,2);"hits=";HIT

  110R=7-RND(14)

  120IF R<0 THEN S=3-R: E=38

  130IF R>=0 THEN S=3: E=38-R

  140IF DIR=-1 THEN T=S: S=E: E=T

  150FOR I=S TO E STEP DIR

  160PROCENEMY

  170PROCFIRE

  180IF F=1 THEN PROCGUIDE

  190NEXT I

  200PRINT TAB(X(I),Y(I+R));" ";

  210IF F=1 THEN FIN=1

  220IF FIN=2 THEN A=11: GOTO 300

  230PROCSTREN

  240IF F=1 THEN F=0: PRINT TAB(MX,MY);CHR$(226):SOUND 0,-15,5,5:FOR Q=1 TO 1000:NEXT Q

  250PRINT TAB(MX,MY);" ";

  260MX=19:MY=23

  270PRINT TAB(MX,MY);CHR$(225)

  280IF FIN=0 THEN GOTO 150

  290FIN=0

  300NEXT A

  310IF FIN=2 THEN PRINT TAB(0,29);"They got you"

  320PRINT TAB(21,2);"hits=";HIT

  330PRINT TAB(0,30);"You hit ";HIT

  340INPUT "Another game",A$

  350IF A$="Y" THEN RUN

  360*FX 4,0

  370END

  380DEF PROCENEMY

  390PRINT TAB(X(I),Y(I+R));" ";

  400PRINT TAB(X(I+DIR),Y(I+R+DIR));CHR$(224);

  410IF RND(1)<.5 THEN ENDPROC

  420IF C<=20 AND MX-X(I+DIR)=0 THEN FIN=2:GOTO 840

  430MOVE X(I+DIR)*32+16,1023-Y(I+R+DIR)*32-32

  440D=300-RND(600)

  450PLOT 1,D,-250

  460SOUND 1,-15,100,1

  470MOVE X(I+DIR)*32+16,1023-Y(I+R+DIR)*32-32

  480PLOT 2,D,-250

  490ENDPROC

  500DEF PROCFIRE

  510IF F=1 THEN ENDPROC

  520A$=INKEY$(0)

  530*FX 15,1

  540IF A$="" THEN ENDPROC

  550PRINT TAB(MX,MY);" ";

  560IF ASC(A$)=&88 THEN MX=MX-8: GOTO 590

  570IF ASC(A$)=&89 THEN MX=MX+8: GOTO 590

  580ENDPROC

  590PRINT TAB(MX,MY);CHR$(225);

  600F=1

  610ENDPROC

  620DEF PROCSTREN

  630C=0

  640J=1023-19*32-16

  650FOR I=17*32 TO 24*32  STEP 4

  660C=C+POINT(I,J)

  670NEXT I

  680C=INT(C/48*100)

  690PRINT TAB(0,0);"Shield Strength ";C;" % "

  700ENDPROC

  710DEF PROCGUIDE

  720PRINT TAB(MX,MY);" ";

  730MY=MY-1

  740IF MY<3 THEN F=0: FIN=1: GOTO 860

  750PRINT TAB(MX,MY);CHR$(225);

  760IF MX<>X(I+DIR) THEN ENDPROC

  770IF MY-Y(I+R+DIR)>2 OR MY-Y(I+R+DIR)<0 THEN ENDPROC

  780FIN=1

  790PRINT TAB(MX,MY);" ";

  800MX=X(I+DIR)

  810MY=Y(I+R+DIR)

  820HIT=HIT+1

  830F=0

  840MOVE X(I+DIR)*32+16,1023-Y(I+R+DIR)*32-32

  850PLOT 1,0,Y(I+R+DIR)*32-MY*32

  860PRINT TAB(MX,MY);CHR$(226);

  870SOUND 0,-15,5,5

  880FOR Q=1 TO 1000:NEXT Q

  890IF FIN=2 THEN GOTO 920

  900PRINT TAB(X(I+DIR),Y(I+R+DIR));" ";

  910PRINT TAB(X(I+DIR),Y(I+R+DIR)+1);" ";

  920I=E+DIR

  930ENDPROC

  940DEF PROCPATH

  950DIM X(50),Y(50)

  960X=0: Y=0

  970N=39

  980FOR I=1 TO N

  990X=X+1

 1000Y=16-INT (((20-X)*(20-X))/25)

 1010X(I)=X

 1020Y(I)=Y

 1030NEXT I

 1040I=1

 1050HIT=0

 1060ENDPROC

 1070DEF PROCBLOCK

 1080COLOUR 129

 1090FOR I=0 TO 1

 1100PRINT TAB(17,20-I);SPC(6);

 1110NEXT I

 1120MX=19

 1130MY=23

 1140COLOUR 128

 1150COLOUR 3

 1160PRINT TAB(MX,MY);CHR$(225)

 1170F=0

 1180FIN=0

 1190ENDPROC

 1200DEF PROCINIT

 1210VDU 23,224,&DD,&DD,&DD,&FF,&FF,&3C,&18,&18

 1220VDU 23,225,&18,&3C,&7E,&18,&18,&3C,&7E,&E7

 1230VDU 23,226,&24,&24,&4F,&4A,&34,&70,&4A,&D4

 1240VDU 19,0,4,0,0,0

 1250VDU 19,1,0,0,0,0

 1260VDU 19,2,7,0,0,0

 1270VDU 19,3,3,0,0,0

 1280VDU 23,1;0;0;0;0

 1290COLOUR 128

 1300CLS

 1310*FX 4,1

 1320ENDPROC

Nine Hole Golf
This is a colour graphics game that combines both driving and putting and even includes the hazard of bunkers. You play around a nine hole course with two stages at each hole - the fairway and the green. When you RUN it, notice how, in the first stage, the golfer makes his swing and how the ball flies through the air.

How to play

At the start of each hole you are told the distance to the hole - marked on the screen by a flag - and asked to select which club you wish to use. If you've ever played golf or watched it on TV you'll know that the lower the number of the club the further it will drive the ball. In other words, select the 1 iron to drive a long way and the 8 iron for the really close shots. If you land in a bunker you'll find that your next shot is not as effective as it normally would be. If you overshoot the green you'll get a new go at the hole and if you drive the ball off the screen you forfeit the hole and move on to the next one. Otherwise, once you get close enough to the hole you'll find yourself on the green. A message will tell you how far you have to putt to the hole and will ask you to select the appropriate club. If you overshoot while putting you will find yourself still at some distance from the hole, and you will have to carry on putting until your ball drops into the hole. Your score for each hole is displayed at the end of each hole and a score card for all nine holes is displayed at the end of each round.

Typing tips

As well as the three user-defined graphics characters, you will also find a capital "O" used in this program in line 1400. It marks the hole on the putting green.

Subroutine structure


20
Defines graphics characters


80
Initialises variables


180
Sets up and plays each hole


670
Reports score for each hole


750
End of game


790
Plots balls' flight


1100
Lost ball routine


1160
Displays swinging club


1320
Putting routine


1600
Ball in bunker routine

Scope for improvement
You may have noticed that the score card at the end of the game does not total your score nor compare it with any ideal par for the course. You might like to add both these features. You will need to play the game a few times to discover what figure to set as the par.

Program

   10REM Nine Hole Golf

   20REM Flag

   30VDU 23,224,&08,&0C,&0E,&08,&08,&08,&08,&3E

   40REM golfer

   50VDU 23,225,&18,&3C,&5A,&3C,&18,&24,&24,&42

   60REM bunker

   70VDU 23,226,&18,&3E,&FE,&FF,&7F,&3E,&0E,&0C

   80DIM T(9)

   90B=1: XH=0:  XC=0

  100YH=0:  HT=0:  YC=0

  110MODE 1

  120VDU 19,0,2,0,0,0

  130VDU 19,1,0,0,0,0

  140VDU 19,2,3,0,0,0

  150VDU 19,3,0,0,0,0

  160VDU 23,1;0;0;0;0

  170VDU 5

  180FOR H=1 TO 9

  190CLS

  200PRINT TAB(10,0);"Hole number ";H

  210VDU 5

  220FOR Z=1 TO 8

  230XB=RND(300)+600

  240YB=RND(300)+600

  250GCOL 0,2

  260MOVE XB,YB

  270PRINT CHR$(226)

  280NEXT Z

  290REM Drive section

  300X=RND(100)

  310Y=RND(200)+350

  320XT=RND(300)+800

  330YT=RND(200)+800

  340D=SGN (XT-X)*SQR ((XT-X)*(XT-X)+(YT-Y)*(YT-Y))

  350GCOL 0,1

  360MOVE XT,YT

  370PRINT CHR$(224);

  380GCOL 0,3

  390MOVE X,Y

  400PRINT CHR$(225)

  410VDU 4

  420PRINT TAB(0,25);"Distance to next hole is ";INT(D);SPC(3)

  430INPUT"which club (1 to 8)",C

  440IF C<1 OR C>8 THEN GOTO 430

  450C=INT ((9-C)/B)+1

  460VDU 5

  470PROCSWING

  480GOSUB 790

  490B=1

  500D=SGN (XT-XHT)*SQR ((XT-XHT)*(XT-XHT)+(YT-YHT)*(YT-YHT))

  510VDU 4

  520IF D<-50 THEN PRINT TAB(1,20);"You overshot-try another hole":FOR Q=1 TO 1000:NEXT Q:GOTO 190

  530IF D<50 THEN PRINT TAB(1,20);"on the green": GOTO 1320

  540FOR Q=1 TO 500:NEXT Q

  550VDU 5

  560GCOL 4,4

  570MOVE XHT,YHT

  580PRINT "."

  590MOVE X,Y

  600PRINT CHR$(225)

  610X=XHT

  620Y=YHT

  630GOTO 350

  640PRINT TAB(10,2);"You took ";T(H);" strokes"

  650FOR Q=1 TO 1000:NEXT Q

  660NEXT H

  670CLS

  680PRINT TAB(10,2);"This round"

  690PRINT

  700FOR I=1 TO 9

  710PRINT TAB(8);"Hole";I;

  720IF T(I)=-1 THEN PRINT " lost ball":GOTO 740

  730PRINT " ";T(I);" strokes"

  740NEXT I

  750PRINT TAB(0,25);

  760INPUT "Another round ",A$

  770IF LEFT$(A$,1)="Y" THEN RUN

  780VDU 20:CLS:END

  790REM HIT ROUTINE

  800VT=C*(8+RND(1))

  810HT=0

  820XH=0

  830REM plot ball

  840Q=((YT-Y)/(XT-X))

  850VV=VT*(SIN (45*PI/180))

  860XC=(X+16)

  870YC=Y-16

  880VH=VT*(COS (45*PI/180))

  890HT=HT+VV

  900YH=Q*XH

  910VV=VV-8

  920XH=XH+VH

  930YH=Q*XH

  940IF XH+XC>1280 THEN  YH=0: YT=0: YC=0: XH=0: XC=0: GOTO 1100

  950IF YH+HT+YC>1024 THEN YH=0: YT=0: YC=0: XH=0: XC=0: GOTO 1100

  960IF HT<=0 THEN GOTO 1040

  970GCOL 4,4

  980MOVE XH+XC,YH+HT+YC

  990PRINT "."

 1000FOR Z=1 TO 50:NEXT Z

 1010MOVE XH+XC,YH+HT+YC

 1020PRINT "."

 1030GOTO 890

 1040XHT=XH+XC

 1050YHT=YH+HT+YC

 1060IF POINT(XH+XC+16,YH+HT+YC-24)=2 THEN GOTO 1600

 1070MOVE XH+XC,YH+HT+YC

 1080PRINT "."

 1090RETURN

 1100VDU 4

 1110PRINT TAB(6,1);"You've lost your ball !!"

 1120SOUND 1,-15,50,4

 1130T(H)=-1

 1140FOR Q=1 TO 1000:NEXT Q

 1150GOTO 660

 1160DEF PROCSWING

 1170T(H)=T(H)+1

 1180XS=X+16

 1190YS=Y-10

 1200FOR S=-5 TO -40 STEP -5

 1210A=S/30*PI

 1220SX=30*SIN A:  SY=30*COS A

 1230GCOL 4,4

 1240PLOT 4,XS,YS

 1250PLOT 1,SX,SY

 1260IF S<>-30 THEN FOR Q=1 TO 10:NEXT Q

 1270IF S=-30 THEN SOUND 1,-5,50,2

 1280PLOT 4,XS,YS

 1290PLOT 1,SX,SY

 1300NEXT S

 1310ENDPROC

 1320REM putting

 1330CLS

 1340XG=RND(5)

 1350YG=15

 1360XH=RND(15)+10

 1370YH=15

 1380D=XH-XG

 1390IF D<0 THEN  D=ABS (D)

 1400PRINT TAB(XH,YH);"O";

 1410PRINT TAB(XG,YG); CHR$(225);

 1420PRINT TAB(1,18); "DISTANCE TO HOLE IS ";D;SPC(3)

 1430INPUT "WHICH CLUB (1 TO 8)",C

 1440IF C<1 OR C>8 THEN GOTO 1430

 1450T(H)=T(H)+1

 1460H1=8-C+RND(2)

 1470D=D-H1

 1480FOR Z=XG+1 TO XG+H1

 1490PRINT TAB(Z,YH);".";

 1500FOR Q=1 TO 200:NEXT Q

 1510PRINT TAB(Z,YH);" ";

 1520NEXT Z

 1530PRINT TAB(XG,YG);" ";

 1540XG=XG+H1

 1550IF XG=XH THEN GOTO 1580

 1560IF D<0 THEN CLS:  XG=XH+D: GOTO 1380

 1570GOTO 1400

 1580REM in the hole

 1590GOTO 640

 1600REM in the bunker

 1610VDU 4

 1620GCOL 0,3

 1630PRINT TAB(1,28);"in the bunker";

 1640SOUND 1,-15,50,5

 1650FOR Q=1 TO 2000:NEXT Q

 1660B=2

 1670PRINT TAB(1,28);SPC(20)

 1680RETURN

Noughts and Crosses

Noughts and crosses is a perennial favourite because it is a simple game of strategy. The problem with playing it against a computer is that the computer can be programmed so that the person challenging it can never win. However, this program makes your Electron an opponent who can be beaten. The Electron will make sensible moves, but it is not infallible so it is worth playing on until you beat it. It's actually a very good way of learning about game-playing strategy.

How to play

This game is played on a simple three-by-three grid in the traditional way. You have the 'X' and play first. To make your move you have to specify which square to place your mark on. Type in the row number first, then the column number. For example, type 11 to place your nought in the top, left-hand corner. If you type a number in the wrong format (for example 1,1) or a number that does not correspond to a position on the gird - for example, 41 - the Electron won't accept it and will beep at you. If you type the number of a position that is already occupied, a message to that effect will be displayed. Once you've made your move, the computer replies with its O and you make your next move. At the end, the Electron will display "I WIN" if it has been successful, "YOU WIN" if you've been successful and "DRAW" if it's stalemate. The board has to be completely filled for the game to be over.

Typing tips

A capital O is used in this program - look out for it in the print statement in line 960.

Subroutine structure


20
Main play loop


120
Evaluates computer's move


630
Tries each move


770
Gets player's move


920
Displays moves


1020
Sets up screen and plots frame


1220
End of game

Programming details

The method used for the computer to play noughts and crosses is based on an advanced technique from artificial intelligence. The program only looks one move ahead when deciding its move - in other words it does not try to take account of the next move you'll make - which is why it slips up sometimes and allows you to win!

Program

   10REM  Noughts and Crosses

   20MODE 5

   30PROCINIT

   40PROCMOVE

   50PROCBOARD

   60PROCREPLY

   70IF FIN=1 THEN GOTO 1260

   80IF FIN=2 THEN PROCBOARD :GOTO 1220

   90IF DR=1 THEN GOTO 1240

  100PROCBOARD

  110GOTO 40

  120DEF PROCVAL

  130FOR Z=1 TO 4

  140X(Z)=0

  150Y(Z)=0

  160NEXT Z

  170FOR L=1 TO 3

  180S=0

  190T=0

  200FOR K=1 TO 3

  210IF A(L,K)=1 THEN S=S+1

  220IF B(L,K)=1 THEN T=T+1

  230NEXT K

  240IF S=0 THEN Y(T+1)=Y(T+1)+1

  250IF T=0 THEN X(S+1)=X(S+1)+1

  260NEXT L

  270FOR L=1 TO 3

  280T=0

  290S=0

  300FOR K=1 TO 3

  310IF A(K,L)=1 THEN S=S+1

  320IF B(K,L)=1 THEN T=T+1

  330NEXT K

  340IF S=0 THEN Y(T+1)=Y(T+1)+1

  350IF T=0 THEN X(S+1)=X(S+1)+1

  360NEXT L

  370PROCDIA1

  380PROCDIA2

  390IF X(4)=1 THEN FIN=1:ENDPROC

  400IF Y(4)=1 THEN FIN=2

  410E=128*Y(4)-63*X(3)+31*Y(3)-15*X(2)+7*Y(2)

  420ENDPROC

  430DEF PROCDIA1

  440T=0

  450S=0

  460FOR K=1 TO 3

  470T=T+A(K,K)

  480S=S+B(K,K)

  490NEXT K

  500IF S=0 THEN X(T+1)=X(T+1)+1

  510IF T=0 THEN Y(S+1)=Y(S+1)+1

  520ENDPROC

  530DEF PROCDIA2

  540T=0

  550S=0

  560FOR K=1 TO 3

  570T=T+A(4-K,K)

  580S=S+B(4-K,K)

  590NEXT K

  600IF S=0 THEN X(T+1)=X(T+1)+1

  610IF T=0 THEN Y(S+1)=Y(S+1)+1

  620ENDPROC

  630DEF PROCREPLY

  640M=-256: DR=1

  650FOR J=1 TO 3

  660FOR I=1 TO 3

  670IF A(I,J)=1 OR B(I,J)=1 THEN GOTO 730

  680DR=0: B(I,J)=1

  690PROCVAL

  700IF FIN=1 THEN ENDPROC

  710IF E>M THEN M=E: A=I:B=J

  720B(I,J)=0

  730NEXT I

  740NEXT J

  750B(A,B)=1

  760ENDPROC

  770DEF PROCMOVE

  780PRINT TAB(0,25);

  790INPUT "Your move (row col) ",A$

  800IF LEN(A$)<>2 THEN SOUND 1,-10,100,2:GOTO 780

  810J=VAL(MID$(A$,1,1)):I=VAL(MID$(A$,2,1))

  820IF I<1 OR I>3 THEN SOUND 1,-10,50,2:GOTO 780

  830IF J<1 OR J>3 THEN SOUND 1,-10,50,2:GOTO 780

  840IF A(I,J)=1 THEN GOTO 890

  850IF B(I,J)=1 THEN GOTO 890

  860A(I,J)=1

  870PRINT TAB(0,25);SPC(100)

  880ENDPROC

  890PRINT TAB(0,28);"Position already"'"occupied"

  900SOUND 1,-10,100,2

  910GOTO 780

  920DEF PROCBOARD

  930FOR J=1 TO 3

  940FOR I=1 TO 3

  950IF A(I,J)=1 THEN PRINT TAB(I*3+2,J*6);"X";

  960IF B(I,J)=1 THEN PRINT TAB(I*3+2,J*6);"O";

  970IF A(I,J)+B(I,J)=0 THEN PRINT TAB(I*3+2,J*6);" ";

  980NEXT I

  990PRINT

 1000NEXT J

 1010ENDPROC

 1020DEF PROCINIT

 1030DIM A(3,3)

 1040DIM B(3,3)

 1050DIM X(4),Y(4)

 1060PROCBOARD

 1070VDU 23,1;0;0;0;0

 1080GCOL 0,3

 1090PLOT 4,450,900

 1100PLOT 1,0,-600

 1110PLOT 4,650,900

 1120PLOT 1,0,-600

 1130PLOT 4,250,525

 1140PLOT 1,600,0

 1150PLOT 4,250,725

 1160PLOT 1,600,0

 1170FIN=0

 1180DR=0

 1190COLOUR 3

 1200COLOUR 128

 1210ENDPROC

 1220PRINT TAB(0,30);"I WIN"

 1230GOTO 1270

 1240PRINT TAB(0,30);"DRAW"

 1250GOTO 1270

 1260PRINT TAB(0,30);"YOU WIN"

 1270INPUT "Another game Y/N",A$

 1280IF A$="Y" THEN RUN

 1290CLS

Fruit Machine

Here's a way of playing the fruit machine without spending a penny. Your Electron gives you 100 pence to start with, takes 10 pence for every go, and awards you a sum between 5 pence and 50 pence every time you come up with a winning combination. You can give up while you are winning or carrying on playing until you are broke.

Although short to type in, this program includes some really clever graphics techniques so that you see the drum of the fruit machine rotate smoothly, using only BASIC. In addition, there are sound effects that signal winning combinations. So, listen out for the jackpot!

How to play

There are four symbols in the display - cherries, banana, apple and bell. All the winning combinations are displayed on the screen while you play. These combinations are winners whether they occur on the line and not just as in the pattern suggested by the screen display. Notice, however, that where blanks occur you need some symbol other than the same type. To play, just RUN and then answer "Y" every time you want another spin. If you do not answer "Y" then the computer will tell you how much money you are taking home with you. Once you run out of money the game is over.

Subroutine structure

20
Initialisation


50
Main play loop


170
Sets starting points of drum


220
Spins drum


400
Pay out routine


470
Defines graphics characters and displays title frame


740
Jackpot routine


800
Signals when broke

Programming details

This program uses some very tricky programming techniques - which is why it achieves its effect in so short a length of BASIC. The patterns for each of the shapes are stored in an array, one line of dots to each array element. Each time the fruit machines drum is printed, a different section of the array is used to load the user-defined graphics. You can think of the section of the array that is used as being defined by a window which moves down by one row of dots each time the characters are printed. This produces the visual illusion of a smoothly rotating drum.

Scope for improvement
If you like adding graphics and sound effects to programs there is scope in this game. For example, you could include a surround that looks like a one-armed bandit and sounds of cascading coins and the drum rotating.

Program

   10REM Fruit Machine

   20MODE 5

   30PROCINIT

   40M=100

   50PROCRES

   60M=M-10

   70PROCSPIN

   80PROCPAY

   90IF M<=0 THEN PROCBUST

  100PRINT TAB(0,29);"You have ";M;" p";SPC(3)

  110PRINT "Another spin ?"

  120A$=INKEY$(0)

  130IF A$="" THEN GOTO 120

  140IF A$="Y" THEN GOTO 50

  150PRINT TAB(0,31);"You take home ";M;" p"

  160END

  170DEF PROCRES

  180X=(RND(4)-1)*10+1

  190Y=(RND(4)-1)*10+1

  200Z=(RND(4)-1)*10+1

  210ENDPROC

  220DEF PROCSPIN

  230S=RND(2)+2

  240FOR I=0 TO S*10

  250VDU 23,229,C(X),C(X),C(X+1),C(X+1),C(X+2),C(X+2),C(X+3),C(X+3)

  260VDU 23,232,C(X+4),C(X+4),C(X+5),C(X+5),C(X+6),C(X+6),C(X+7),C(X+7)

  270VDU 23,230,C(Y),C(Y),C(Y+1),C(Y+1),C(Y+2),C(Y+2),C(Y+3),C(Y+3)

  280VDU 23,233,C(Y+4),C(Y+4),C(Y+5),C(Y+5),C(Y+6),C(Y+6),C(Y+7),C(Y+7)

  290VDU 23,231,C(Z),C(Z),C(Z+1),C(Z+1),C(Z+2),C(Z+2),C(Z+3),C(Z+3)

  300VDU 23,234,C(Z+4),C(Z+4),C(Z+5),C(Z+5),C(Z+6),C(Z+6),C(Z+7),C(Z+7)

  310PRINT TAB(7,20);CHR$(229);" ";CHR$(230);" ";CHR$(231)

  320PRINT TAB(7,21);CHR$(232);" ";CHR$(233);" ";CHR$(234)

  330IF X=40 THEN X=0

  340IF Y=40 THEN Y=0

  350IF Z=40 THEN Z=0

  360X=X+1:Y=Y+1:Z=Z+1

  370NEXT I

  380X=X-1:Y=Y-1:Z=Z-1

  390ENDPROC

  400DEF PROCPAY

  410REM calculates winnings

  420IF X=1 AND Y=1 AND Z=1 THEN PROCJACK:ENDPROC

  430IF (X=31)+(Y=31)+(Z=31)=-2 THEN M=M+10:SOUND 1,-15,200,5

  440IF (X=21)+(Y=21)+(Z=21)=-2 THEN M=M+25:SOUND 1,-15,200,5

  450IF (X=1)+(Y=1)+(Z=1)=-1 THEN M=M+5:SOUND 1,-15,200,5:ENDPROC

  460ENDPROC

  470DEF PROCINIT

  480VDU 23,224,&06,&0A,&14,&24,&44,&CF,&EF,&E6

  490VDU 23,225,&02,&0C,&1C,&38,&38,&1C,&0C,&02

  500VDU 23,226,&18,&3C,&3C,&3C,&7E,&FF,&18,&18

  510VDU 23,227,&0C,&18,&7A,&FF,&FF,&FF,&7E,&3C

  520VDU 23,228,&00,&00,&00,&7E,&7E,&00,&00,&00

  530PRINT TAB(0,5);CHR$(224),CHR$(225),CHR$(226),CHR$(227),CHR$(228)

  540CLS

  550PRINT TAB(5);"F R U I T"

  560PRINT TAB(4);"M A C H I N E"

  570PRINT ''"YOU HAVE `1.00"'"TO GAMBLE"

  580PRINT '"EACH SPIN COSTS 10p"

  590PRINT 'CHR$(224);" ";CHR$(224);" ";CHR$(224);" WINS 50p"

  600PRINT 'CHR$(226);" ";CHR$(226);" - WINS 25p"

  610PRINT '"- ";CHR$(227);" ";CHR$(227);" WINS 10p"

  620PRINT'CHR$(224);" - - WINS 5p"

  630DATA &06,&0A,&14,&24,&44,&CF,&EF,&E6,0,0

  640DATA &02,&0C,&1C,&38,&38,&1C,&0C,&02,0,0

  650DATA &18,&3C,&3C,&3C,&7E,&FF,&18,&18,0,0

  660DATA &0C,&18,&7A,&FF,&FF,&FF,&7E,&3C,0,0

  670DATA &06,&0A,&14,&24,&44,&CF,&EF,&E6,0,0

  680DIM C(48)

  690FOR I=1 TO 48

  700READ C(I)

  710NEXT I

  720VDU 23,1;0;0;0;0

  730ENDPROC

  740DEF PROCJACK

  750FOR I=50 TO 200 STEP 8

  760SOUND 1,-15,I,10

  770NEXT I

  780M=M+50

  790ENDPROC

  800DEF PROCBUST

  810PRINT TAB(0,30);"YOU ARE BROKE !!!!!!!!!!!!"

  820END

Rainbow Squash

This is a colourful version of the popular computer squash game which is also enhanced by the addition of sound - a cheery beep every time the ball bounces either on the sides of the court or against the bat and a dismal tone every time a ball goes out of play. Its other feature is that as you improve in skill the game gets more difficult, and if you then start to get worse it gets easier. This means that your Electron will always give you a challenge that is suited to your ability - which makes it the perfect partner.

How to play

At the start of the game the bat is at the bottom of the screen and in the centre. You control the left and right movement of the bat by pressing the appropriate arrow keys. Every time you make two hits in succession the position of the bat changes - it moves nearer to the top of the screen - which makes returning the ball more difficult. If you then miss a shot the ball will move back one position, making it easier. You score a point for every hit and you will be served a total of ten balls. Information about the number of balls played and hits scored is displayed on the screen continuously.

Subroutine structure


20
Sets up screen and colours and initialise variables


150
Defines graphics characters


180
Sets up initial screen display


200
Main play loop


450
Draws court


620
Bounce routine


770
Moves bat up screen


890
End of game - prints final score, offers another game and re-runs or restores screen display


1010
Function to move bat


1070
Determines colour of bat

Programming details

Although this program is fairly short it uses some clever tricks to ensure that the old positions of the bat and ball are blanked out in the correct colours. Writing a program of this complexity is much easier using procedure calls and function definitions. Notice the use of PROCCOL which changes the background colour and the use of FNBAT which controls the bat's movement. The definition of this function includes *FX 15,1 - which flushes the buffer after every key press. This means that the bat responds instantaneously to the player's change of direction.

Program

   10REM Rainbow Squash

   20MODE 1

   30VDU 19,1,0,0,0,0

   40VDU 19,0,6,0,0,0

   50VDU 19,2,2,0,0,0

   60VDU 19,3,3,0,0,0

   70*FX 4,1

   80*FX 11,1

   90*FX 12,1

  100H=0

  110HT=0

  120D=19

  130BALL=0

  140C=2

  150VDU 23,224,&FF,&FF,&FF,&FF,&FF,&FF,&FF,&FF

  160VDU 23,225,&3C,&7E,&FF,&FF,&FF,&FF,&7E,&3C

  170VDU 23,1;0;0;0;0

  180PROCCOURT

  190X=10

  200BALL=BALL+1

  210IF BALL>10 THEN GOTO 890

  220A=10+RND(6)

  230B=1

  240V=1

  250W=1

  260Y=D

  270COLOUR 128:COLOUR 1

  280 PRINT TAB(10,21);"BALL ";BALL;

  290X=FNBAT(X):PROCCOL

  300PRINT TAB(X,Y);CHR$(E);CHR$(224)CHR$(224)CHR$(224);CHR$(F);

  310COLOUR 128:COLOUR 1

  320PRINT TAB(20,21);"HIT ";HT

  330REM bounce ball

  340PROCBOUNCE

  350COLOUR 1:COLOUR 128

  360IF B+W<>Y THEN Y=D:GOTO 290

  370SOUND 1,-15,50,10

  380PROCCOL

  390PRINT TAB(X+1,Y);SPC(3);

  400PROCBOUNCE

  410PRINT TAB(A,B);" "

  420IF D<19 THEN D=D+1

  430H=0

  440GOTO 200

  450DEF PROCCOURT

  460CLS

  470COLOUR 1

  480FOR I=0 TO 39

  490PRINT CHR$(224);

  500NEXT

  510COLOUR 130

  520PRINT STRING$(200," ");

  530PRINT STRING$(80," ");

  540COLOUR 131

  550PRINT STRING$(200," ");

  560PRINT STRING$(80," ");

  570COLOUR 1

  580FOR I=0 TO 30

  590PRINT TAB(0,I);STRING$(8,CHR$(224));TAB(31,I);STRING$(9,CHR$(224));

  600NEXT I

  610ENDPROC

  620DEF PROCBOUNCE

  630COLOUR 128

  640IF B<15 THEN COLOUR 131

  650IF B<8  THEN COLOUR 130

  660PRINT TAB(A,B);" "

  670A=A+V

  680B=B+W

  690IF A=30 OR A=8 THEN V=-V:SOUND 1,-4,89,5

  700IF B=1 THEN W=-W:SOUND 1,-15,89,5

  710IF B+W=Y THEN GOTO 770

  720COLOUR 128

  730IF B<15  THEN COLOUR 131

  740IF B<8  THEN COLOUR 130

  750PRINT TAB(A,B);CHR$(225)

  760ENDPROC

  770R=A-X

  780IF R<1 OR R>3 THEN GOTO 750

  790W=-W

  800SOUND 1,-15,89,5

  810H=H+1

  820HT=HT+1

  830IF H<>1 THEN GOTO 720

  840H=0

  850D=D-1

  860PROCCOL

  870PRINT TAB(X+1,Y);SPC(3);

  880GOTO 720

  890FOR Z=1 TO 500:NEXT Z

  900CLS

  910PRINT TAB(10,10);"You Scored ";HT

  920PRINT TAB(10,15);

  930*FX 15,1

  940*FX 4,0

  950*FX 12,0

  960INPUT "ANOTHER GAME Y/N ",A$

  970A$=LEFT$(A$,1)

  980IF A$="Y" THEN RUN

  990CLS

 1000END

 1010DEF FNBAT(Q)

 1020K=INKEY(0)

 1030*FX 15,1

 1040IF K=&89  AND Q<27 THEN Q=Q+1

 1050IF K=&88 AND Q>7 THEN Q=Q-1

 1060=Q

 1070DEF PROCCOL

 1080COLOUR 128

 1090IF Y<15 THEN COLOUR 131

 1100IF Y<8 THEN COLOUR 130

 1110IF X=27 THEN F=224 ELSE F=32

 1120IF X=7 THEN E=224 ELSE E=32

 1130ENDPROC

Smalltalker

Do you ever find yourself talking to your Electron? Well, if you do you may be disappointed that it never answers back. This program, however, changes all that and gives your Electron the chance to have a conversation with you. Although it may not be able to rival the agony aunts of the glossy magazines, your Electron is anxious to hear about your problems - and has some comments to offer.

Coping with the syntax of the English language is one of the very complicated problems with which this program has to contend. Programs like this one have been developed to extend our knowledge of how language words and how humans identify the key components of conversations. While these serious purposes are usually the province of artificial intelligence it is possible to have a good deal of fun trying to conduct a dialogue with your Electron.

How to use this program
The computer opens each conversation in the same way - by inviting you to tell it your problems. You can give any reply that you wish to and after a few moments' delay your Electron will respond. Try to say more than just "YES" or "NO" when you make further responses but, equally, don't say too much at a time. If you type about a lineful each time, you ought to be able to keep a reasonable conversation going.

Typing tips

Do remember to use upper-case only when entering this program and when using it. As the computer has to match your sentences against its vocabulary, it is also very important to be careful about your spelling. If you type in either the initial program or subsequent responses with misspellings, the computer won't recognise your messages and you won't receive any sensible answers. The apostrophe is the only punctuation mark that should be used in dialogues with the Electron.

This is by far the longest program in this collection and it is an especially trying one to type in because of all its long lines and frequent repetition. If you want a short cut to using it, remember that there are cassette tapes available.

Subroutine structure


20
Main program loop


100
Initial message and set up


400
Input human's sentence


560
Divides sentences into words


650
Changes tense/pronouns


820
Tense/pronouns data


920
Finds keywords in sentence


1070
Keywords data


1250
Keyword responses


2210
Prints computer's response


2270
Requests sensible input

Programming details

This program works by taking a sentence and splitting it down into individual words and the responding according to a list of keywords that it searches for in each sentence. So if, for example, your sentence contains the word 'why', the response 'Some questions are difficult to answer' will always be given by the computer. When the computer fails to find a specify reply to a sentence, one of a number of responses is selected at random.

Although this technique sounds simple, the actual details of the program are really quite tricky as, amongst other things, the computer has to deal with tense changes and with the syntax of pronouns. It is therefore quite a difficult program to write or to modify extensively. Equally, despite the apparent simplicity of its underlying technique, it succeeds in making plausible responses on a surprising number of occasions.

Scope for improvement
If you wish to add to the list of keywords that the computer recognises, you need to notice how, in subroutine 1070, the keywords are paired with the line number of the subroutine that responds to them. Also it is important to be aware of the priorities assigned to each keyword. If two keywords are present in a sentence then the one first in the list will be acted upon.

Program

   10REM Smalltalker

   20MODE 4:DIM W(20,2)

   30GOSUB 100

   40GOSUB 400

   50GOSUB 560

   60GOSUB 920

   70IF NM<>0 THEN ON NM GOSUB 1250,1270,1290,1310,1730,1780,1690,1710,1610,1380,2140,1330,1470,1900,2040,1800,1820,1840,1330,1880,1860

   80GOSUB 2210

   90GOTO 40

  100CLS

  110PRINT TAB(10);"HI"

  120PRINT

  130PRINT "I WOULD LIKE YOU TO TALK TO ME"

  140PRINT "BUT I DON'T HAVE EARS SO WILL"

  150PRINT "YOU TYPE SENTENCES ON MY KEYBOARD"

  160PRINT "IN UPPER CASE"

  170PRINT

  180PRINT "DON'T USE ANY PUNCTUATION APART"

  190PRINT "FROM APOSTROPHIES WHICH ARE IMPORTANT"

  200PRINT

  210PRINT

  220PRINT "WHEN YOU HAVE FINISHED TYPING"

  230PRINT "PRESS RETURN"

  240PRINT '' "TELL ME YOUR PROBLEMS"

  250R$=""

  260M$=""

  270D$=""

  280DIM N$(3)

  290N$(1)="PLEASE GO ON"

  300N$(2)="I'M NOT SURE I UNDERSTAND YOU"

  310N$(3)="TELL ME MORE"

  320DIM I$(3)

  330I$(1)="LET'S TALK SOME MORE ABOUT YOUR"

  340I$(2)="EARLIER YOU SPOKE OF YOUR"

  350I$(3)="DOES THAT HAVE ANYTHING TO DO WITH YOUR"

  360DIM J$(2)

  370J$(1)="ARE YOU JUST BEING NEGATIVE"

  380J$(2)="I SEE"

  390RETURN

  400A$=""

  410B$=INKEY$(0)

  420REM Input section

  430IF B$="" THEN GOTO 410

  440IF INKEY$(0)<>"" THEN GOTO 440

  450IF ASC(B$)=13 THEN GOTO 510

  460IF ASC(B$)=&7F AND A$<>"" THEN A$=LEFT$(A$,LEN(A$)-1):GOTO 490

  470IF ASC(B$)<32 OR ASC(B$)>126 THEN GOTO 400

  480A$=A$+B$

  490PRINT TAB(0,30);A$;" "

  500GOTO 410

  510IF A$=R$ THEN PRINT TAB(0,31);"YOU'RE REPEATING YOURSELF":PRINT:PRINT:GOTO 400

  520R$=A$

  530IF A$="" THEN GOTO 400

  540A$=" "+A$

  550RETURN

  560REM Divides sentences into words

  570N=1

  580B=0

  590FOR I=1 TO LEN(A$)

  600IF ((MID$(A$,I,1)=" " OR MID$(A$,I,1)=",") AND B=0) THEN B=1

  610IF ((MID$(A$,I,1)<>" " AND MID$(A$,I,1)<>",") AND B<=1) THEN W(N,1)=I:B=2

  620IF ((MID$(A$,I,1)=" " OR MID$(A$,I,1)=",") AND B=2) THEN W(N,2)=I-1:N=N+1:B=0

  630NEXT I

  640W(N,2)=LEN(A$)

  650FOR I=1 TO N

  660RESTORE

  670READ B$

  680IF B$="S" THEN GOTO 800

  690IF B$<>MID$(A$,(W(I,1)),(W(I,2)-W(I,1)+1)) THEN GOTO 780

  700READ C$

  710A$=LEFT$(A$,(W(I,1)-1))+C$+RIGHT$(A$,(LEN(A$)-W(I,2)))

  720W(I,2)=W(I,2)+LEN(C$)-LEN(B$)

  730FOR J=I+1 TO N

  740W(J,2)=W(J,2)+LEN(C$)-LEN(B$)

  750W(J,1)=W(J,1)+LEN(C$)-LEN(B$)

  760NEXT J

  770GOTO 800

  780READ B$

  790GOTO 670

  800NEXT I

  810RETURN

  820DATA "MY","YOUR*","I","YOU*"

  830DATA "MUM","MOTHER","DAD","FATHER"

  840DATA"DREAMS","DREAM","YOU","I*","ME","YOU*"

  850DATA "YOUR","MY*","MYSELF","YOURSELF*"

  860DATA "YOURSELF","MYSELF*","I'M","YOU'RE*"

  870DATA "YOU'RE","I'M","AM","ARE*"

  880DATA "I'M","YOU'RE*"

  890DATA "WERE","WAS"

  900DATA "ARE","AM"

  910DATA "S","S"

  920RESTORE:FOR I=1 TO 34:READ Q$:NEXT I

  930READ B$,NM

  940IF B$="S" THEN GOTO 1010

  950I=1

  960IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))<>B$ THEN GOTO 990

  970T$=RIGHT$(A$,LEN(A$)-(W(I,2)))

  980RETURN

  990I=I+1:IF I<=N THEN GOTO 960

 1000GOTO 930

 1010NM=0

 1020IF M$<>"" THEN GOTO 1050

 1030P$=N$(RND(3))

 1040RETURN

 1050P$=I$(RND(3))+M$

 1060RETURN

 1070DATA "COMPUTER",1,"MACHINE",1,"PROGRAM",1

 1080DATA "LIKE",2,"SAME",2,"ALIKE",2

 1090DATA "IF",3,"EVERYBODY",4

 1100DATA "CAN",5,"CERTAINLY",6

 1110DATA "HOW",7,"BECAUSE",8

 1120DATA "ALWAYS",9

 1130DATA "EVERYONE",4,"NOBODY",4

 1140DATA "WAS",10

 1150DATA "I*",11

 1160DATA "NO",12

 1170DATA "YOUR*",13

 1180DATA "YOU'RE*",14,"YOU*",15

 1190DATA "HELLO",16,"MAYBE",17

 1200DATA "MY*",18,"NO",19

 1210DATA "YES",6,"WHY",20

 1220DATA "PERHAPS",17,"SORRY",21

 1230DATA "WHAT",20

 1240DATA "S",0

 1250P$="DO COMPUTER WORRY YOU ?"

 1260RETURN

 1270P$="IN WHAT WAY ?"

 1280RETURN

 1290P$="WHY TALK OF POSSIBILITIES"

 1300RETURN

 1310P$="REALLY "+B$+" ?"

 1320RETURN

 1330IF I=N THEN GOTO 1360

 1340I=I+1

 1350IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="ONE" THEN B$=B$+" ONE":GOTO 1310

 1360P$=J$(RND(2))

 1370RETURN

 1380IF I=N THEN GOTO 2270

 1390I=I+1

 1400IF I>N THEN GOTO 1020

 1410IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))<>"YOU*" THEN GOTO 1440

 1420P$="WHAT IF YOU WERE "+RIGHT$(A$,(LEN(A$)-W(I,2)))+" ?"

 1430RETURN

 1440IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))<>"I*" THEN GOTO 1020

 1450P$="WOULD YOU LIKE TO BELIEVE I WAS "+RIGHT$(A$,(LEN(A$)-(W(I,1)+1)))

 1460RETURN

 1470I=I+1

 1480IF I>N THEN GOTO 1360

 1490IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="MOTHER" THEN GOTO 1590

 1500IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="FATHER" THEN GOTO 1590

 1510IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="SISTER" THEN GOTO 1590

 1520IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="BROTHER" THEN GOTO 1590

 1530IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="WIFE" THEN GOTO 1590

 1540IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="HUSBAND" THEN GOTO 1590

 1550IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="CHILDREN" THEN GOTO 1590

 1560IF LEN(T$)>10 THEN M$=T$

 1570P$="YOUR "+T$+" ?"

 1580RETURN

 1590P$="TELL ME MORE ABOUT YOUR FAMILY"

 1600RETURN

 1610P$="GIVE ME A PARTICULAR EXAMPLE"

 1620RETURN

 1630I=I+1

 1640IF I>N THEN P$="AM I WHAT ?":RETURN

 1650P$="WHY ARE YOU INTERESTED IN WHETHER I AM "+RIGHT$(A$,W(I,1)+" OR NOT ?"

 1660RETURN

 1670P$="DO YOU THINK YOU ARE "+RIGHT$(A$,LEN(A$)-W(I,2))

 1680RETURN

 1690P$="WHAT DO YOU ASK ?"

 1700RETURN

 1710P$="TELL ME ABOUT ANY OTHER REASONS"

 1720RETURN

 1730I=I+1

 1740IF I>N THEN P$="WHAT ?":RETURN

 1750IF MID$(A$,W(I,1),(W(I,2)-W(I,1))+1)="I*" THEN P$="DO YOU BELIEVE I CAN "+RIGHT$(A$,(LEN(A$)-W(I,2)))+" ?":RETURN

 1760IF MID$(A$,W(I,1),(W(I,2)-W(I,1))+1)="YOU*" THEN P$="DO YOU BELIEVE YOU CAN "+RIGHT$(A$,LEN(A$)-W(I,2))+" ?":RETURN

 1770GOTO 1010

 1780P$="YOU SEEM VERY POSITIVE"

 1790RETURN

 1800P$="PLEASED TO MEET YOU - LET'S TALK ABOUT YOUR PROBLEMS"

 1810RETURN

 1820P$="COULD YOU TRY TO BE MORE POSITIVE"

 1830RETURN

 1840P$="WHY ARE YOU CONCERNED ABOUT ME "+T$

 1850RETURN

 1860P$="YOU DON'T HAVE TO APOLOGISE TO  ME"

 1870RETURN

 1880P$="SOME QUESTIONS ARE DIFFICULT TO ANSWER..."

 1890RETURN

 1900I=I+1

 1910IF I>N THEN GOTO 1020

 1920P$="I AM SORRY TO HEAR THAT YOU ARE "+MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))

 1930IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="SAD" THEN RETURN

 1940IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="UNHAPPY" THEN RETURN

 1950IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="DEPRESSED" THEN RETURN

 1960IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="SICK" THEN RETURN

 1970P$="HOW HAVE I HELPED YOU TO BE "+MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))

 1980IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="HAPPY" THEN RETURN

 1990IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="ELATED" THEN RETURN

 2000IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="GLAD" THEN RETURN

 2010IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="BETTER" THEN RETURN

 2020P$="IS IT BECAUSE YOU ARE "+MID$(A$,W(I,1),W(I,2)-W(I,1)+1)+" YOU WOULD LIKE TO TALK TO ME ?"

 2030RETURN

 2040IF I=1 THEN GOTO 2060

 2050IF MID$(A$,W(I-1,1),(W(I-1,2)-W(I-1,1)+1))="ARE*" THEN GOTO 1670

 2060I=I+1

 2070IF I>N THEN GOTO 2270

 2080IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="ARE*" THEN GOTO 1900

 2090IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="WANT" OR MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="NEED" THEN P$="WHAT WOULD IT MEAN IF YOU GOT "+RIGHT$(A$,LEN(A$)-W(I,2)):RETURN

 2100IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="THINK" THEN P$="DO YOU REALLY THINK SO":RETURN

 2110IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="CAN'T" OR MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="CANNOT" THEN P$="HOW DO YOU KNOW YOU CAN'T"+RIGHT$(A$,LEN(A$)-W(I,2)):RETURN

 2120IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="FEEL" THEN P$="TELL ME MORE ABOUT HOW YOU FEEL":RETURN

 2130GOTO 1020

 2140IF I-1<1 THEN GOTO 2160

 2150IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="AM" THEN GOTO 1630

 2160I=I+1

 2170IF I>=N THEN P$="WHAT AM I ?":RETURN

 2180IF MID$(A$,W(I,1),(W(I,2)-W(I,1)+1))="AM" THEN P$="WHY DO YOU THINK SO ?":RETURN

 2190P$="IS THAT WHAT YOU THINK OF ME ?":RETURN

 2200RETURN

 2210FOR J=1 TO LEN(P$)

 2220IF MID$(P$,J,1)="*" THEN GOTO 2240

 2230COLOUR 0:COLOUR 129:PRINT MID$(P$,J,1);:COLOUR 1:COLOUR 128

 2240NEXT J

 2250PRINT '''

 2260RETURN

 2270P$="PLEASE TALK SENSIBLY !"

 2280RETURN

